
INTRODUCCIÓN

Todos quienes nos hemos sentado a programar seriamente en un lenguaje orientado a WINDOWS, tuvimos que haber escuchado la palabra API (Interfaz de programación de aplicaciones), mucho de nosotros en nuestra programación rutinaria las usamos algunas veces sin entenderla y sin conocerlas, es decir, copiamos la función de una página WEB, la pegamos en Visual Basic y listo. Solo sabemos que esa función me retorna un valor y ya!. El entender las API es mucho más grande y complejo que lo que vemos en las páginas WEB.

Las API de WINDOWS son funciones del sistema operativo, WINDOWS como un sistema completamente abierto nos permite usar sus API para adaptar nuestro SOFTWARE a él, recuerden que WINDOWS es quién gobierna a nivel de SOFTWARE, y todo software se tiene que adaptar a ese gobierno, vean las API como las leyes de ese gobierno para diseñar un SOFTWARE y si no respetas las leyes, el SOFTWARE no servirá.

En este manual voy explicar las API de Windows con el lenguaje Visual Basic, aunque no es el lenguaje apropiado. Ya que lo que vamos a ver, los controles de VB (Visual Basic), lo resumen notablemente. El lenguaje ideal, para aquellos que quieren entrar más a fondo en este mundo es C++ o VC++, estos lenguaje trabajan con las API directamente, más que todo C++, ya que VC++ contiene las clases MFC (Microsoft Fundamention Classes), pero les aseguro que si no entienden las API no entenderán las clases MFC.

Tengo que decir que este manual es para personas de conocimientos MEDIOS-AVANZADOS en VB. Esta manual tiene como finalidad explicar en una forma general el funcionamiento de las funciones API, vamos a mostrar ejemplos de API, pero no vamos a cubrir todas las API ya que eso sería demasiado extenso. Y cabe decir que yo no conozco todas las API de Windows, ya que para que se imaginen estamos hablando de aproximadamente de 800 API, 35.000 Constantes, etc.

PRINCIPIO DE LAS API

Algo que es difícil de ver y de entender, pero que tiene que estar bien claro es que todo lo que vemos es una “ventana”. En este momento ustedes se preguntarán ¿Como es eso?, ¿Es un botón una ventana?, ¿Es un combobox una ventana?, pues la respuesta es sí.

Imagínense por separado  una ventana (“FORM o FORMULARIO”) y un botón. A la ventana quítenle la barra superior que tiene el CAPTION y los botones de MINIMIZAR, RESTAURAR y CERRAR. Entonces nos queda un fondo y un borde, ahora póngale a la ventana (“FORM”), un fondo gris y pónganle el mismo tamaño que el botón, y le ponemos una etiqueta en el centro que diga “BOTON”, si se lo están imaginando, van haber que es EXACTAMENTE lo mismo, no hay diferencia. Es decir, todos los controles son ventanas pero con diferentes ESTILOS. WINDOWS identifica a esas ventanas con lo que él llama HANDLE (HWND) o lo que vemos en VB como el atributo  HWND de los controles. Justificación o prueba de lo que digo es que todos los controles en VB tiene la propiedad HWND.

HWND o HANDLE

El HWND es un número UNICO que se le da a la ventana cuando ella se ejecuta (Hablo de ventana y me refiero a todo, controles, formularios, etc) . Por ejemplo si abriéramos el procesador de texto WORD y la hoja de cálculo EXCEL. Windows le da un número, por ejemplo:

EXCEL = 300001

WORD = 300002

Este número sirve para que Windows sepa a que programa manda un mensaje, o a que programa se le da prioridad ,etc. Este número no es fijo, puede cambiar cada vez que abramos el mismo programa.  Pero el número si es UNICO, es decir no hay dos 300001, por que imagínense si hubiera dos programas con el mismo número o HWND. Si Windows envía un mensaje “X” al HWND 300001, si hay dos programas  con el mismo número PUFF! Colapso!. Veamos el HWND como un índice de ventanas abiertas (Repito cuando me refiero a ventana me refiero a todo).

HDC (Device Context) o Dispositivo de Contexto

El HDC es el área gráfica de HWND (A partir de ahora verán abreviaturas de algunas palabras como por ejemplo VB Visual Basic, HWND handler y HDC Device Context). Esta área es donde vamos a dibujar y colorear el HWND que seleccionemos. Imagínense a un pintor, un pintor para pintar usa un caballete y un papel o lienzo. Bueno el caballete es el HWND y el papel o lienzo es el HDC. Esto téngalo muy en claro por que el HDC es muy usado.

MENSAJES

En esta sección vamos a ver nuestra primera API, que es el SendMessage, como todos sabemos la acción que usa un OBJETO para comunicarse con otro OBJETO es el MENSAJE, ese mensaje lleva una información que se procesará en el OBJETO receptor.

El API se compone de esta manera:

SendMessage (hwnd,wMsg,wParam, lParam)

Estudiemos su estructura:

hwnd = es ese número que le asigna WINDOW a ese objeto o ventana receptora.

wMsg = es el Mensaje.

wParam = es la información  que se le envía al hwnd.

lParam = es otra información que también se le envía al hwnd.

En esta estructura uno de los puntos más importantes es el wMsg. Imaginen por un momento, lo siguiente. Ustedes quieren saludar a una persona. Sería algo así:

SendMessage(NOMBREDEPERSONA, SALUDO, “BUENOS DIAS”, NULL)

NOMBREDEPERSONA es quien recibe el mensaje.

SALUDO = Es el tipo de mensaje que se envía

“BUENOS DIAS” = Es la información que se envía.

En Windows los tipos de mensajes vienen definidos por CONSTANTES. Existen miles de constantes y tipos de mensaje que se pueden enviar a un objeto o hwnd. Ahora hay ciertos mensaje que están definidos para cierto tipo de ventana. Por ejemplo hay mensajes solo para TEXTBOX, otras solo para LISTBOX, etc.

Veamos nuestro primer programa en Visual Basic. Para este ejemplo es necesario tener un botón en un formulario. A continuación copien el siguiente código:

Private Declare Function SendMessage Lib "user32" Alias "SendMessageA" (ByVal hwnd As Long, ByVal wMsg As Long, ByVal wParam As Long, lParam As Any) As Long

Const WM_SETTEXT = &HC

Private Sub Command1_Click()

Dim Resultado As Long

Resultado = SendMessage(Command1.hwnd, WM_SETTEXT, 0, ByVal "GRACIAS")

End Sub

Private Sub Form_Load()

    Command1.Caption = "PRESIONAME"

End Sub

ANALISIS

Private Declare Function SendMessage Lib "user32" Alias "SendMessageA" (ByVal hwnd As Long, ByVal wMsg As Long, ByVal wParam As Long, lParam As Any) As Long

Cuando se declara un API en un formulario deben de colocarle la palabra PRIVATE antes de DECLARE sino VB les indicará un error. En caso de poner la función en un modulo (.bas) ahí iría sin el PRIVATE.

Lib “user32”, esta es la librería en donde se encuentra la función API, normalmente las funciones API’S dependiendo del tipo de funciones están almacenadas en distintas librerías. Por ejemplo las gráficas están en gdi32.dll, etc. De todas maneras en las páginas WEB, libros, etc. Cuando declaran el API ahí verán la librería a la cual pertenece.

El ALIAS no es materia de este manual y tampoco es necesario para lo que vamos haber.

Van a notar en los parámetros que se le pasan a la función, que tienen la palabra ByVal, esto significa que los parámetros se pasan por valor y no por referencia. He de suponer que ustedes a este nivel tienen claro esos términos.

El “lParam as Any” el Any es que acepta cualquier objeto en este caso una cadena “GRACIAS”.

Con este código verán que cuando le dan click al botón, cambia y dice “GRACIAS”, ahora que sucedió:

Resultado = SendMessage(Command1.hwnd, WM_SETTEXT, 0, ByVal "GRACIAS")

RESULTADO = Todos los mensaje o en su mayoría retornan VALORES, siempre todas las funciones API regresan un número de tipo LONG. En este caso no interesa el valor a retornar pero más adelante veremos que sí.


HWND = COMMAND1.hwnd

Este es nuestro objeto receptor, es decir el objeto a quién le estamos enviando el mensaje.


MENSAJE = WM_SETTEXT

Mensaje que declaramos como constante en el área de “DECLARACIONES”, normalmente estos son valores hexadecimales.


Wparam = 0 

En esta caso aquí no va el mensaje.


LParam = “ ByVal “GRACIAS”


El ByVal es muy importante ya que obliga a que se pase el valor por copia y no por referencia. Y este es el mensaje que le pasamos.

MAS SOBRE MENSAJES.

Si ustedes cambian el Command1.hwnd y colocan Me.hwnd y lo corren cuando aprieten el botón van a ver el GRACIAS, no en el botón sino en el CAPTION del formulario, esto es por que  el mensaje WM_SETTEXT sirve para cualquier hwnd. Ahora como saber que mensaje sirve para tal objeto.

Bueno aquí te muestro una lista de algunos de ellos:


BM_



Button mensaje


CB_



Combo Box Mensaje


CDM_



Common Dialog Mensaje


DRV_



Driver Mensaje


EM_



Edit control o TextBox Mensaje


LB_



List Box Mensaje


WM_



General Window Mensaje


Etc.

Cuando ustedes vean un mensaje o constante que empiece por BM_ significa que es un mensaje para un botón si lo usan en un Combo Box no funcionará. Es por esa razón que WM_SETTTEXT funcionó tanto para el botón como para el formulario.

Tanto para los mensajes como para la funciones tienen que tomar en cuenta un detalle, vean que el mensaje es WM_SETTEXT el SET se usa cuando asignamos y el GET para tomar un valor, es decir, cuando vean una función:


SETXNOMBRE = significa que vamos asignar algo.


GETXNOMBRE = significa que vamos a obtener algo.

En este caso el WM_ = mensaje que identifica que es un hwnd o ventana, SET = asignamos, TEXT = texto.

En este caso el TEXTO de la VENTANA botón es lo que llamamos en VB la propiedad CAPTION, al igual con el formulario.

Ahora vamos a probar algo interesante pero con GET

Declara este mensaje 

Const WM_GETTEXT = &HD

Luego en el evento click del botón cambiamos el código que esta por el siguiente:

Private Sub Command1_Click()

Dim Resultado As Long

Resultado = SendMessage(Command1.hwnd, WM_GETTEXT, 0, 0)

End Sub

Arranca y presiona el botón verás que no cambia nada, ni se ve nada, tu dirás bueno el Texto del hwnd debe de estar en la variable “Resultado”, bueno vamos a probar pon debajo de:

    Resultado = SendMessage(Command1.hwnd, WM_GETTEXT, 0, 0)

   Msgbox Resultado

Cuando lo corras verás que te muestra 0, las funciones API de WINDOW retornan nada mas resultados de tipo LONG, no retornan ni STRING, ni FLOAT, etc. Ustedes dirán ¿bueno y como hacemos?. Cuando un API retorna una información ya sea de texto o una estructura (ya veremos eso mas adelante) se usan de esta manera:

Cambia el código en el evento Click del botón por este:

Private Sub Command1_Click()

Dim Resultado As Long

Dim Texto As String

Texto = Space$(20)

Resultado = SendMessage(Command1.hwnd, WM_GETTEXT, 15, ByVal Texto)

MsgBox Texto

End Sub


Dim Texto as String

Verás que declaramos una variable de tipo String

Texto = Space$(20)

Le asignamos 20 espacios en blancos a la variable, cuando quieres llenar una variable de tipo STRING sea cual sea la función API, tienes que asignarle el espacio en memoria, eso lo hacemos asignándole 20 espacios en blanco, pero el  número puede variar.

SendMessage(Command1.hwnd, WM_GETTEXT, 15, ByVal Texto)

En esta función lo que cambia es el mensaje y los parámetros, el mensaje cambia a GETTTEXT es decir, “tomar el texto”, wParam = 15 esto significa que le dices a WINDOW que te regrese 15 caracteres nada más, si la cadena a regresar es mayor a 15 te le regresa incompleta. lParam = ByVal Texto; “Texto” es la variable donde se almacenará el Texto que Windows regrese. Si tu declaras por ejemplo:

Texto = Space$(2)

Cuando Windows intente llenar la variable no lo va llenar con el texto completo y te puede aparecer caracteres extraños. Esto es por que cuando quiere grabar el Texto del hwnd en la memoria se encuentra con una variable que fue inicializada con dos caracteres. 

Normalmente se declaran las variables con 255, ejemplo, Texto = Space$(255), luego si quieres eliminar los espacios en blancos usamos la función TRIM de VB para ello.

El SendMessage es una de las API mas usadas en la programación bajo WINDOW, bueno es obvio que nosotros nunca vamos hacer lo que hicimos arriba para asignarles un Texto a una hwnd (Verán que uso la palabra hwnd y no ventana esto es por que no quiero que se confundan de que ventana es el Formulario, necesito que se acostumbre a que un hwnd sea cual fuera, es una ventana). Esto es por que los controles de VB resumen ese paso en sus propiedades, es decir, cuando cambias el texto a un hwnd (botón) lo que haces es usar su propiedad CAPTION. Obviamente ya ustedes se imaginarán lo que hace CAPTION en su interior.

La finalidad de explicar todo lo que dije arriba, fue para que vean como se maneja un API, ya que las API’S en su manejo son similares dependiendo de su tipo. A partir de ahora vamos a empezar a ver API’S un poco más complejas, ya que entenderlas implica entender a WINDOW y eso es un poco complicado. VB es un lenguaje muy sencillo y tal vez necesiten de las API’S para problemas extremos, pero les aseguro que si quieren aprender lenguajes como VC++ necesitarán del mas mínimo conocimiento de API’S para entenderlo.

SENDMESSAGE Y POSTMESSAGE

PostMessage es exactamente igual a SendMessage, la diferencia radica en que cuando mandan mensaje con SendMessage el código que esta inferior a él no se ejecuta hasta que termine, por ejemplo:

Private Sub Command1_Click()

    Dim Resultado As Long

    Dim Texto As String

    Texto = Space$(20)

    Resultado = SendMessage(Command1.hwnd, WM_GETTEXT, 15, ByVal Texto)

    MsgBox Texto ‘ Esta línea no se ejecuta hasta que el SendMessage se ejecute por completo

End Sub

Si usamos PostMessage:

Private Sub Command1_Click()

    Dim Resultado As Long

    Dim Texto As String

    Texto = Space$(20)

    Resultado = PostMessage(Command1.hwnd, WM_GETTEXT, 15, ByVal Texto)

    MsgBox Texto ‘ Esta línea se ejecuta sin necesidad de que PostMessage haya finalizado.    

End Sub

El uso de PostMessage no es muy usual, inclusive todavía no he visto un ejemplo práctico con esta función. Pero ya saben de que existe.

NOTA: Ustedes se preguntaran, ¿COMO SABER CUANDO USAR WPARAM Y LPARAM? o ¿COMO SE QUE VALOR LES PASO?, para ello no existe ninguna regla, cuando ustedes quieran usar un mensaje con SendMessage ustedes tienen que buscar en la ayuda de MSDN que parámetros se le envían a ese mensaje.

WINDOWS MENSSAGE

En esta sección vamos a ver algunos mensajes útiles para las ventanas o hwnd usando el SendMessage, es importante mencionar, que no voy a explicar todos los mensajes para cada control, sino los que yo considero más importantes, recuerden que esta manual no esta orientado, para ser un diccionario de API sino explicar como funcionan mediante ejemplos.

EDIT CONTROL O TEXTBOX

· Mensaje : EM_GETLINE

Valor : Const EM_GETLINE = &HC4

Mensaje que funciona en un Textbox Multiline, este permite tomar el Texto que se encuentra en una línea específica.

Dim Respuesta As Long

Dim Texto As String

Texto = Space$(10)

Respuesta = SendMessage(Text1.hwnd, EM_GETLINE, 1, ByVal Texto)

MsgBox Texto

Lo único que hay que explicar es el valor que toma wParam que es 1, este es el número de línea de la cual queremos ver el texto. Si ponemos 0 seria la primera línea, 1 la segunda línea, así sucesivamente.

· Mensaje: EM_GETLINECOUNT

Valor: Const EM_GETLINECOUNT = &HBA

Mensaje que sirve para retornar la cantidad de líneas que posee un Textbox configurado de manera multiline.

Dim Respuesta As Long

    Respuesta = SendMessage(Text1.hwnd, EM_GETLINECOUNT, 0, 0)

    MsgBox Respuesta

Si vemos aquí, si nos interesa el valor de retorno del Mensaje, verán que no se le pasa ningún parámetro o información, con el tipo de Mensaje basta para que él sepa a que nos referimos. El número de líneas se almacena en la variable Respuesta.

· Mensaje: EM_GETMODIFY

Valor: Const EM_GETMODIFY = &HB8

Esta mensaje sirve para saber si un TEXTBOX, multiline o simple, se ha modificado. 

Dim Respuesta As Long

 Respuesta = SendMessage(Text1.hwnd, EM_GETMODIFY, 0, 0)

       MsgBox Respuesta

Los valores que retornan es 0 si no se ha modificado y 1 si fue modificado.

· Mensaje : EM_LINELENGTH

Valor: Const EM_LINELENGTH = &HC1

Este mensaje sirve para tomar la longitud de una línea de un TEXTBOX sea multiline o no.

    Dim Respuesta As Long

    Respuesta = SendMessage(Text1.hwnd, EM_LINELENGTH, 1, 0)

    MsgBox Respuesta

Aquí el mensaje retorna la longitud sería como el LEN de VB. wParam es el número de la línea de la cual queremos tomar su longitud. Esto si el Textbox es Multiline, en caso de no serlo no importa el número que pongas en wParam, siempre te retornará la longitud de la única línea que tiene.

· Mensaje : EM_SETSEL 
Valor: Const EM_SETSEL = &HB1

Esta mensaje sirve para seleccionar una porción de la cadena que esta contenida en el TEXTBOX.

SendMessage Text1.hwnd, EM_SETSEL, 5, 1

MsgBox Text1.SelText

Aquí wParam recibe la posición donde se inicia la selección, el texto seleccionado se encontrara en la Propiedad .SelText del Textbox. 

Noten que esta función la invocamos diferente. Cuando el valor que retorna una función no interesa, podemos usar la función como que si llamáramos a un método, Por ejemplo:

Respuesta = SendMessage(Text1.hwnd, EM_LINELENGTH, 1, 0)

En este caso nos interesa el resultado de la función por que la variable “Respuesta” toma el valor de la longitud de la cadena.

Respuesta = SendMessage(Text1.hwnd, EM_SETSEL, 5, 1)

Aquí “respuesta” no nos sirve para nada, por consiguiente, cuando el valor de la función sea cual sea el API no nos interese, podemos usarla de esta manera:

SendMessage Text1.hwnd, EM_SETSEL, 5, 1

Siendo completamente válido esta instrucción. Pero tienen que tomar en  cuenta que algunas veces lo que retorna una función no lo usamos con un propósito, pero algunas veces el resultado que devuelve una función puede estar diciendo si la operación se ejecuto o no se ejecuto con éxito. Les repito eso dependerá del mensaje que ustedes utilicen.

COMBO BOX CONTROL

· Mensaje: CB_DIR

Valor : Const CB_DIR = &H145

Este mensaje, el cual lo veo muy importante permite llenar un COMBO BOX con información del sistema. En este caso a wParam le pasaremos unos parámetros ya predefinidos por WINDOW.

Parámetros que se pasan por Wparam:

Const DDL_ARCHIVE = &H20

Const DDL_DIRECTORY = &H10

Const DDL_DRIVES = &H4000

Const DDL_EXCLUSIVE = &H8000

Const DDL_HIDDEN = &H2

Const DDL_READONLY = &H1

Const DDL_READWRITE = &H0

Const DDL_SYSTEM = &H4

Ejemplo:

Dim path As String    

path = App.path

SendMessage Combo1.hwnd, CB_DIR, DDL_DRIVES, ByVal path

Se darán cuenta que al ejecutarlo, todas las letras y unidades de la computadora se almacenan en el combo, verán que en wParam  toma el parámetro que declaramos como constante DDL_DRIVES, los parámetros también pueden ser constantes ya establecidas o que poseen un valor. En lParam verán que es la ruta de nuestro programa, que es lo que exige el mensaje como segundo parámetro. Para saber que se les pasa a los otros parámetros que van en wParam, les recomiendo que busquen la ayuda en Línea de MSDN, ya que la ayuda normal, no muestra que parámetros son.

· Mensaje : CB_FINDSTRING

Valor: Const CB_FINDSTRING = &H14C

Este mensaje muy importante en el momento de hacer búsqueda en un COMBO ya que permite localizar cierto registro en el COMBO. En este mensaje wParam es el índice en donde empezamos a buscar y lParam es la cadena que buscamos.

    Dim Texto As String

    Dim  Respuesta As Long

    Texto = "Prueba"

    Respuesta = SendMessage(Combo1.hwnd, CB_FINDSTRING, -1, ByVal Texto)

    MsgBox Respuesta

Podremos ver que wParam le coloque el valor de –1, eso es para que busque en toda la lista del combo, si pongo por ejemplo 2 en wParam, empezará a buscar desde el  registro 3, recuerden que el combo empieza en el índice 0.

Si la función encuentra la cadena regresa el índice en donde se encuentra en el combo, en caso de que no exista regresa –1. En esta función la búsqueda no es exacta, es decir:

“PRUEBA” Imagínense que tienen esta cadena en el Combo, si ustedes asignan la variable “Texto” de esta manera:


Texto = “Pru”

Cuando realice la búsqueda para el tipo de mensaje “PRUEBA” es igual a “PRU”, esto es por que  la cadena en el combo empieza por “PRU”. Para hacer una búsqueda exacta vean el mensaje a continuación.

· Mensaje : CB_FINDSTRINGEXACT

Valor : Const CB_FINDSTRINGEXACT = &H158

Esta mensaje es igual al anterior con la diferencia que busca la cadena exacta en la lista del combo.

    Dim Texto As String

    Dim Respuesta As Long

    Texto = "Prueba"

    Respuesta = SendMessage(Combo1.hwnd, CB_FINDSTRINGEXACT, -1, ByVal Texto)

    MsgBox Respuesta

Si seguimos el ejemplo de “PRU”, para este mensaje “PRU” no es igual a “PRUEBA”, ya que esta función busca la cadena que sea EXACTAMENTE igual.

Estas dos funciones son muy importantes en el momento de realizar búsquedas en COMBO que contengan mucha información, es común de los programadores en VB hacer:


For i = 0 to Combo1.listcount –1



Código para buscar


Next

Usando el SendMessage les aseguro que van a tener resultados increíblemente rápidos.

· Mensaje : CB_SHOWDROPDOWN

Valor : Const CB_SHOWDROPDOWN = &H14F

Esta función se utiliza para mostrar la lista desplegable del combo box.


SendMessage Combo1.hwnd, CB_SHOWDROPDOWN, True, 0

En este caso el valor de wParam toma el valor TRUE para mostrar la lista y FALSE para esconderla.

GRÁFICOS CON API


Tal vez una de los aspectos mas complicados de la API son lo gráficos, esto se debe a que para entender la API’S gráficas hay que entender como funciona Windows a nivel gráfico.

Todo sabemos, por ejemplo, que si queremos poner en un formulario de VB de color AZUL lo que hacemos es buscar la propiedad BackColor, le asignamos un color y listo. Bueno en las API no es tan simple como suena. Para explicarlo de una forma más didáctica, voy a usar el ejemplo de un PINTOR. 

Imagínense a un PINTOR que va pintar una CASA, la dueña de la casa decide que sea todo el exterior de color AZUL, pero quiere en el interior que se dibuje unas líneas de color AZUL CLARO,  algo loco pero es un ejemplo. Bueno que hace el pintor, para pintar el interior hace las siguientes pasos.

1. Agarra una BROCHA (BRUSH).

2. Le asigna un color, en este caso el color AZUL

3. Y luego procede a pintar.

4. Cuando termina se limpia la brocha.

Ahora los pasos para pintar las líneas serían:

1. Busca un PINCEL (PEN)

2. Le asignamos color, y tipo de PINCEL.

3. Procedemos a pintar las líneas.

4. Cuando termina se limpia el PINCEL.

Eso seria algo Básico, ahora vamos a ver todo esto en la programación. Primero pondré el ejemplo y luego haré la explicación de las API’S que se usan.

Para la ejecución de este ejemplo es necesario colocar el control TIMER en un formulario y su propiedad INTERVAL en 10. A continuación copien el siguiente código.

Private Declare Function CreateSolidBrush Lib "gdi32" (ByVal crColor As Long) As Long

Private Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Private Declare Function FillRect Lib "user32" (ByVal hdc As Long, lpRect As RECT, ByVal hBrush As Long) As Long

Dim Brocha As Long

Private Type RECT

        Left As Long

        Top As Long

        Right As Long

        Bottom As Long

End Type

Dim RECT As RECT

Private Sub Form_Unload(Cancel As Integer)

DeleteObject Brocha

End Sub

Private Sub Timer1_Timer()

    Randomize

    Brocha = CreateSolidBrush(RGB(Int(Rnd * 255) + 1, Int(Rnd * 255) + 1, Int(Rnd * 255) + 1))

    RECT.Left = Int(Rnd * (Screen.Width \ 15)) + 1

    RECT.Top = Int(Rnd * (Screen.Height \ 15)) + 1

    RECT.Right = Int(Rnd * 200) + 1

    RECT.Bottom = Int(Rnd * 200) + 1

    FillRect Me.hdc, RECT, Brocha

End Sub

Ejecútalo y verás que se presentan una serie de rectángulo que cambian de color y de tamaño. Ahora, recuerdan los pasos que hacia un pintor para pintar la pared de una casa vamos a recordarlo.

1. Agarra una BROCHA (BRUSH).

Brocha = CreateSolidBrush(

Uso el API CreateSolidBrush para crear una brocha.

2. Le asigna un color, en este caso el color AZUL

Brocha = CreateSolidBrush(RGB(Int(Rnd * 255) + 1, Int(Rnd * 255) + 1, Int(Rnd * 255) + 1))

En este caso no es nada mas azul, sino que elijo el color aleatoriamente. Para obtener un color usamos la función RGB.

3. Procedo a pintar

FillRect Me.hdc, RECT, Brocha

Uso el API FillRect para pintar un rectángulo con relleno.

4. Cuando termina se limpia la brocha.

DeleteObject Brocha

Esta API la usamos para borrar la brocha de la memoria.

Ahora para pintar líneas coloque el siguiente código:

Private Type POINTAPI

    x As Long

    y As Long

End Type

Const PS_SOLID = 0

Private Declare Function LineTo Lib "gdi32" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long) As Long

Private Declare Function MoveToEx Lib "gdi32" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long, lpPoint As Any) As Long

Private Declare Function CreatePen Lib "gdi32" (ByVal nPenStyle As Long, ByVal nWidth As Long, ByVal crColor As Long) As Long

Private Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Private Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Dim p As POINTAPI  

Private Sub Form_Paint()

        loA = CreatePen(PS_SOLID, 6, RGB(255, 200, 200))

        lonA = SelectObject(Me.hdc, loA)

        MoveToEx Me.hdc, 100, 100, p

        LineTo Me.hdc, 120, 100

        MoveToEx Me.hdc, 180, 100, p

        LineTo Me.hdc, 120, 130

        SelectObject Me.hdc, lonA

        DeleteObject (loA)

End Sub

Ejecútalo y verás a dos líneas, una recta y otra inclinada hacia arriba. Vamos a ejecutar los pasos que indique arriba en el ejemplo del pintor:

1. Busca un PINCEL (PEN)

loA = CreatePen( 

Para crear un pincel utilizamos el API CreatePen

2. Le asignamos color, y tipo de PINCEL.

loA = CreatePen(PS_SOLID, 6, RGB(255, 200, 200)) 

En breve explico los valores de esta API

3. Procedemos a pintar las líneas.

MoveToEx Me.hdc, 100, 100, p

LineTo Me.hdc, 120, 100

MoveToEx Me.hdc, 180, 100, p

LineTo Me.hdc, 120, 130

4. Cuando termina se limpia el PINCEL.

DeleteObject  loA

Ahora voy a explicar con un poco más detalle lo que vimos arriba para que entiendan el funcionamiento gráfico de Windows, hay muchas API de Windows a nivel gráfico, y algunas simplifican el trabajo de otras. Y para explicar un API de seguro utilizaremos otras API para que funcione. Por esa razón en los ejemplos que voy a tratar van ver que cuando explique un API, eso deriva la explicación de otra.

Hay dos API claves las cuales se presentan en muchos ejemplos gráficos, y estas son las funciones API SelectObject y DeleteObject.

SELECTOBJECT

API: Private Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Vemos que entre sus parámetros esta un HDC y un OBJECT. Cuando nosotros creamos un objeto por ejemplo un PINCEL con CreatePen, ese pincel esta en memoria pero no esta asignado a un HDC. Sabemos que un formulario tiene una propiedad HDC que no es nada mas que su dispositivo de contexto o su área gráfica. Esa área gráfica tiene un PEN ya configurado, y si nosotros queremos cambiar ese PEN tenemos que hacer lo siguiente: 

        loA = CreatePen(PS_SOLID, 6, RGB(255, 200, 200))


Aquí creamos el PEN (PINCEL) esta en la memoria

        lonA = SelectObject(Me.hdc, loA)

Aquí con el API SelectObject,  le estamos diciendo, cámbiame en el HDC del formulario el PEN que esta ahí, por “loA” que es nuestro nuevo PEN.

Ahora “lonA” es nuestro PEN viejo o el que tenía el HDC antes de suplantarlo por el nuevo. Cuando nosotros terminamos el trabajo con el PEN que creamos, es recomendable restaurar el PEN que teníamos originalmente. Eso justifica la instrucción que esta más abajo que es:

        SelectObject Me.hdc, lonA

Aquí restauro el PEN que tenía el HDC del formulario originalmente, noten que aquí uso el API como un método debido a que no me interesa el valor que devuelve.

Recuerden, SelectObject se usa para asignar un objeto a un HDC, y el valor que retorna es el objeto del mismo tipo, antiguo o que tenía antes el HDC.

Probablemente o lo mas seguro es que utilicen  SelectObject cuando usen las siguientes API para crear diferentes objetos:

CreateBitmap, CreateBitmapIndirect, CreateCompatibleBitmap, CreateDIBitmap, CreateDIBSection, CreateBrushIndirect, CreateDIBPatternBrush, CreateDIBPatternBrushPt, CreateHatchBrush, CreatePatternBrush, CreateSolidBrush, CreateFont, CreateFontIndirect, CreatePen, CreatePenIndirect, CombineRgn, CreateEllipticRgn, CreateEllipticRgnIndirect, CreatePolygonRgn, CreateRectRgn, CreateRectRgnIndirect

DELETEOBJECT

API: Private Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Esta es mas fácil de explicar, esta API lo que hace es borrar de la memoria el objeto creado.

DeleteObject  loA


De esta manera borramos el PEN que creamos.

Tienen que tener en claro estas dos API ya que ellas se emplean mucho en los ejemplos de API’S a nivel gráfico.

Ustedes se preguntarán ¿Por qué use el API SelectObject en el segundo ejemplo y no el primero en donde use CreateSolidBrush? 

Lo que ocurre aquí es lo siguiente, yo no modifico el HDC del formulario del primer ejemplo por que la API FillRect, posee entre sus parámetros el BRUSH (Brocha), ya el API se encarga de procesar el fondo del rectángulo. No trabajo directamente con el HDC, uso el HDC como papel para dibujar, pero el color de fondo que afecta al Rectángulo ya su API lo procesa automáticamente.

A diferencia con CREATEPEN, cuando uso la API LineTo para dibujar la línea, esta API no tiene entre sus parámetros PEN, eso quiere decir que usa el PEN del HDC donde se dibuje, por esa razón en el segundo ejemplo uso SelectObject y en el primero no.

Las API gráficas tienen muchas formas o diferentes usos, que hacen que parezcan un poco confusa, pero si analizamos un poco y tratamos de entender lo que hacemos estoy seguro que no van a tener problema.

Seguro se preguntaran ¿Por que hacer todo esto?, Así funciona Windows, recuerden que él es el que gobierna y nosotros tenemos que regirnos bajo ese gobierno, el control LINE de VB hace lo mismo que hicimos en el segundo ejemplo de arriba, lo que pasa es que en VB las API se disfrazan en controles. Obviamente que estas API’S no la vamos a usar en nuestro trabajo diario, pero sirven y son claves para entender como Windows trabaja, y son claves también para manejar las API’S que veremos mas adelante.

A continuación voy explicar las API’S restantes que vimos en los ejemplos de arriba:

CREATESOLIDBRUSH

Api: Private Declare Function CreateSolidBrush Lib "gdi32" (ByVal crColor As Long) As Long

Esta API sencilla crea una BROCHA con un color, el cual es su único parámetro, para asignar un color o usamos su valor Hexadecimal o la función RGB de VB. Lo que retorna esta función es la referencia al objeto que creamos. 

Las Brochas o Brush se usan mucho para pintar o rellenar fondos de figuras geométricas o de HDC. Todos hemos visto en el programa PAINT de Windows, un baldecito de pintura que cuando lo pones en el área de pintar y las das click llena toda el área del color seleccionado. Bueno ya se imaginarán como lo hace.

FILLRECT

Api: Private Declare Function FillRect Lib "user32" (ByVal hdc As Long, lpRect As RECT, ByVal hBrush As Long) As Long

Esta API sirve para dibujar un rectángulo con un relleno.

En esta API vemos algo muy importante que no habíamos visto en las API’S anteriores y es que uno de sus parámetros es una ESTRUCTURA. 

El primer parámetro es un HDC, ya todo sabemos que es el HDC, en donde se va a pintar el rectángulo.

El Tercer parámetro es una variable BRUSH, ¿Como sabemos, que es una variable Brush?, las API en el nombre de sus variables usa nombres similares a los objetos que requiere, por ejemplo, hBrush (Obviamente si vemos el nombre de la variable veremos que necesita una BROCHA o BRUSH). 

El segundo parámetro es una estructura RECT.

Vamos a explicar un poco esto de ESTRUCTURA. Algunas API’S de Windows requieren para su funcionamiento un parámetro que en este caso es una ESTRUCTURA, ahora ¿Porque el uso de la ESTRUCTURA?, algunas funciones API’S pueden regresar una cierta cantidad de valores, al igual que asignarle muchos valores. Y los diseñadores de Windows fueron muy astutos en usar la estructura para ello. Imagínense la función API FillRect sin el parámetro RECT. 

Private Declare Function FillRect Lib "user32" (ByVal hdc As Long,x as long, y as long, bottom as long, right as long, ByVal hBrush As Long) As Long

Todo sabemos que un rectángulo tiene una esquina superior llamémosla x,y donde “x” es LEFT y “y” es TOP, a su vez también posee un RIGHT y BOTTOM que no es nada mas que la esquina inferior derecha.

Pueden ver que el API crecería en sus parámetros. En esta API el cambio no sería notable, pero hay algunas API’S que reciben estructuras muy grandes, entonces imagínense una función API donde pasas 20 parámetros; si de por si!, las API son complicadas no me imagino lo mas complicadas que serían si trabajarán de esa manera. Para ello usamos las estructuras. Cuando ustedes vean en un parámetro un tipo de dato que no se asemeja a los tipo de datos tradicionales, téngalo por seguro que eso es una estructura, en este ejemplo se usa la estructura RECT:

Private Type RECT

        Left As Long

        Top As Long

        Right As Long

        Bottom As Long

End Type

Las estructuras son como la definición de un “OBJETO SIMPLE”, “simple” por que nada mas tiene parámetros. Las estructuras no poseen ni métodos, ni funciones. Y se declaran al igual que a un objeto.

Dim RECT As RECT

Luego, antes de llamar al API que la usa le tenemos que añadir los valores correspondiente:

    RECT.Left = Int(Rnd * (Screen.Width \ 15)) + 1

    RECT.Top = Int(Rnd * (Screen.Height \ 15)) + 1

    RECT.Right = Int(Rnd * 200) + 1

    RECT.Bottom = Int(Rnd * 200) + 1

Y ya con las variables inicializadas llamamos al API:

FillRect Me.hdc, RECT, Brocha

En esta función API hay que llenar RECT, antes de llamar a la función, creo que eso es obvio, no podemos pintar un rectángulo si no indicamos adonde va pintado.

Pero en algunos casos que veremos mas adelante, veremos que las estructuras no se llenan antes de invocar a la función API, sino que la misma función API se encarga de llenar cada una de las variables de la estructuras para luego nosotros hacer uso de ellas, eso lo veremos en el API que viene a continuación.

MOVETOEX

Api: Private Declare Function MoveToEx Lib "gdi32" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long, lpPoint As Any) As Long

Esta API la utilizamos para cambiar el punto donde empezamos a dibujar. En el ejemplo anterior usamos el primer MoveToEx de esta manera:


MoveToEx Me.hdc, 100, 100, p 

Lo que hacemos aquí es poner el punto de inicio en la coordenada 100,100 de la pantalla, “p” es una estructura de tipo POINTAPI que se encarga de almacenar las coordenadas en donde estaba puesto el punto de inicio. Cabe decir que como dije arriba aquí “p” no necesita tener valores para que el API funcione, ya que la misma se encarga de llenar la estructura.
LINETO 

Api: Declare Function LineTo Lib "gdi32" Alias "LineTo" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long) As Long

Mas fácil de explicar determina el punto final de la línea, recuerden que una línea básicamente es la unión de dos puntos el punto de inicio lo fijamos con MoveToEx y el punto final lo fijamos con LineTo

LineTo Me.hdc, 120, 100

Aquí los parámetros son simple la X y Y respectivamente del punto final de la línea, estoy seguro que se corren el ejemplo no tendrán ningún problema en entenderlo.

CREATEPEN

Api: Private Declare Function CreatePen Lib "gdi32" (ByVal nPenStyle As Long, ByVal nWidth As Long, ByVal crColor As Long) As Long
Con esta API lo que hacemos es crear nuestra pluma o pincel para dibujar, esta API se compone del parámetro nPenStyle, que es el estilo de como se dibujará la línea, nWidth que no es sino el grosor de la línea y por último crColor que es el color.

        loA = CreatePen(PS_SOLID, 6, RGB(255, 200, 200))

  lonA = SelectObject(Me.hdc, loA)

PS_SOLID, es el estilo de la línea a continuación les presento los demás estilos con sus valores correspondientes:

Private Const PS_DASHDOT = 3         ' _._._._

Private Const PS_DOT = 2           ' .......

Private Const PS_DASH = 1          ' -------

Private Const PS_DASHDOTDOT = 4       ' _.._.._

Private Const PS_INSIDEFRAME = 6

Private Const PS_NULL = 5

Private Const PS_SOLID = 0

Cambien el valor y les dejo a ustedes el trabajo de conocer los diferentes estilos. Luego continuando con el ejemplo, le pasamos 6 al grosor de la línea pero pueden poner el valor que quieran, y bueno el color que de seguro ya lo tienen en claro.

Ya deberían de estar claro por que uso el SelectObject así que me saltare la explicación.

MAS API’S GRÁFICAS

A partir de ahora mostrare en las API’S su declaración y una forma sencilla de usarla ya me saltare explicar con detalles cada una, a menos que lo ameriten.

CREATECOMPATIBLEDC

API : Declare Function CreateCompatibleDC Lib "gdi32" Alias "CreateCompatibleDC" (ByVal hdc As Long) As Long

Esta API se utiliza para crear un nuevo DC en memoria de un DC ya existente.


Ejemplo.


HNewDC = CreateCompatibleDC(Me.HDC)

En ejemplo de arriba creamos una copia del DC del formulario activo en ese momento.

CREATEDC

Api : Declare Function CreateDC Lib "gdi32" Alias "CreateDCA" (ByVal lpDriverName As String, ByVal lpDeviceName As String, ByVal lpOutput As String, lpInitData As DEVMODE) As Long

Se utiliza para crear un DC. Tienen que quedar en claro que el DC es un “Dispositivo de Contexto”, los Dispositivos de contexto que hemos manejado en los ejemplos han sido salidas gráficas a través del monitor o pantalla, pero un DC puede ser enviado a una impresora. En pocas palabras un DC es un dispositivo que puede tener dos salidas o PANTALLA o IMPRESORA. Aunque en los ejemplos usaremos nada mas la salida del Monitor.

Estructura DEVMODE

Public Type DEVMODE

    dmDeviceName As String  CCHDEVICENAME

    dmSpecVersion As Integer

    dmDriverVersion As Integer

    dmSize As Integer

    dmDriverExtra As Integer

    dmFields As Long

    dmOrientation As Integer

    dmPaperSize As Integer

    dmPaperLength As Integer

    dmPaperWidth As Integer

    dmScale As Integer

    dmCopies As Integer

    dmDefaultSource As Integer

    dmPrintQuality As Integer

    dmColor As Integer

    dmDuplex As Integer

    dmYResolution As Integer

    dmTTOption As Integer

    dmCollate As Integer

    dmFormName As String  CCHFORMNAME

    dmUnusedPadding As Integer

    dmBitsPerPel As Integer

    dmPelsWidth As Long

    dmPelsHeight As Long

    dmDisplayFlags As Long

    dmDisplayFrequency As Long

End Type


Esta API tiene 4 parámetros:

· lpDriverName = Determina hacia donde va el dispositivo de contexto, si al DISPLAY o al WINSPOOL.

· LpDeviceName = No lo explico ya que la documentación que he visto para escribir esta manual muestra en su explicación, que usualmente no es usado en WIN32.

· LpOutput = No es usado en aplicaciones WIN32.

· LpInitData = Si lpDriverName es un DISPLAY se coloca vbNullChar. En caso de ser por el WINSPOOL tiene que pasarse la estructura DEVMODE.

Estructura DEVMODE:

· dmDeviceName = Nombre de la salida. Ejm: HP 810, Epson 600, etc.

· DmSpecVersion = Número de versión.

· DmDriverVersion = Número de versión del driver de la impresora.

· DmSize = Tamaño de la estructura

· DmDriveExtra = Usualmente se coloca 0

· DmFields = Usado para decir si la variable esta inicializada.

· DmOrientation = Puede ser DMORIENT_PORTRAIT o DMORIENT_LANDSCAPE.

Public Const DMORIENT_LANDSCAPE = 2

Public Const DMORIENT_PORTRAIT = 1

· dmPaperSize = Tamaño del papel. Valores que puede tomar:

Public Const DMPAPER_10X11 = 45         ' 10 x 11

Public Const DMPAPER_10X14 = 16         ' 10x14

Public Const DMPAPER_11X17 = 17         ' 11x17

Public Const DMPAPER_12X11 = 90

Public Const DMPAPER_15X11 = 46         ' 15 x 11

Public Const DMPAPER_9X11 = 44         ' 9 x 11

Public Const DMPAPER_A2 = 66          ' A2 420 x 594 mm

Public Const DMPAPER_A3 = 8           ' A3 297 x 420 mm

Public Const DMPAPER_A3_EXTRA = 63       ' A3 Extra 322 x 445 mm

Public Const DMPAPER_A3_EXTRA_TRANSVERSE = 68  ' A3 Extra Transverse 322 x 445 mm

Public Const DMPAPER_A3_ROTATED = 76

Public Const DMPAPER_A3_TRANSVERSE = 67     ' A3 Transverse 297 x 420 mm

Public Const DMPAPER_A4 = 9           ' A4 210 x 297 mm

Public Const DMPAPER_A4_EXTRA = 53          ' A4 Extra 9.27 x 12.69

Public Const DMPAPER_A4_PLUS = 60        ' A4 Plus 210 x 330 mm

Public Const DMPAPER_A4_ROTATED = 77

Public Const DMPAPER_A4_TRANSVERSE = 55     ' A4 Transverse 210 x 297 mm

Public Const DMPAPER_A4SMALL = 10        ' A4 Small 210 x 297 mm

Public Const DMPAPER_A5 = 11          ' A5 148 x 210 mm

Public Const DMPAPER_A5_EXTRA = 64       ' A5 Extra 174 x 235 mm

Public Const DMPAPER_A5_ROTATED = 78

Public Const DMPAPER_A5_TRANSVERSE = 61     ' A5 Transverse 148 x 210 mm

Public Const DMPAPER_A6 = 70

Public Const DMPAPER_A6_ROTATED = 83

Public Const DMPAPER_A_PLUS = 57        ' SuperA/SuperA/A4 227 x 356 mm

Public Const DMPAPER_B4_JIS_ROTATED = 79

Public Const DMPAPER_LETTER = 1

Public Const DMPAPER_LETTER_EXTRA = 50       ' Letter Extra 9 \275 x 12

Public Const DMPAPER_LETTER_EXTRA_TRANSVERSE = 56 ' Letter Extra Transverse 9\275 x 12

Public Const DMPAPER_LETTER_PLUS = 59      ' Letter Plus 8.5 x 12.69

Y muchas mas constantes.

· dmPaperLenght = Longitud del dmPaperSize.

· DmPaperWidth = Ancho del dmPaperLenght.

· DmScale = Escala de la salida.

· DmCopies = Número de Copias.

· DmPrintQuality = Calidad o Resolución de la Impresión.

Public Const DMRES_DRAFT = (-1)

Public Const DMRES_HIGH = (-4)

Public Const DMRES_LOW = (-2)

Public Const DMRES_MEDIUM = (-3)

· dmColor = Determina si la imagen se presenta en color o blanco y negro.

Public Const DMCOLOR_COLOR = 2

Public Const DMCOLOR_MONOCHROME = 1

· dmDuplex = Determina si la impresión es Doble-Lado.

Public Const DMDUP_HORIZONTAL = 3

Public Const DMDUP_SIMPLEX = 1

Public Const DMDUP_VERTICAL = 2

· dmYResolution = Punto por Pulgada de el eje y.

· DmtTOption = Fuentes TrueType. 

Public Const DMTT_BITMAP = 1      

Public Const DMTT_DOWNLOAD = 2     

Public Const DMTT_DOWNLOAD_OUTLINE = 4 

Public Const DMTT_SUBDEV = 3      

· dmCollate = Intercalar. Esta propiedad toma o TRUE o FALSE.

· DmFormName = Nombre de la forma. Puede ser Letra o Legal.

· DmUnusedPadding = Se declara en 0

· DmBitsPerPel = Presenta la resolución a nivel de color.

· DmPelsWidth = Representa en ancho de la resolución en píxeles.

· DmPëlsHeight = Determina el alto de la resolución en píxeles.

· DmDisplayFlags = Puede tomar los valores:

Public Const DM_GRAYSCALE = &H1

Public Const DM_INTERLACED = &H2




Y muchos mas.

· dmDisplayFrequency = Frecuencia en Hertz.

Por todo lo que he escribido arriba no se preocupen, yo nada mas he usado esta ESTRUCTURA para obtener la resolución actual y poder cambiar la resolución del sistema. 

Ejm:

Hdc = CreateDC("DISPLAY", vbNullString, vbNullString, ByVal 0&)

GETDC

Api: Declare Function GetDC Lib "user32" Alias "GetDC" (ByVal hwnd As Long) As Long
Obtiene el dispositivo de contexto de un HWND.


Ejm:


Hdc = GetDC(frmName.hwnd)

GETWINDOWDC

Api: Declare Function GetWindowDC Lib "user32" Alias "GetWindowDC" (ByVal hwnd As Long) As Long

Obtiene el HDC, pero completa, es decir, incluyendo Menús, Toolbar, StatuBar, bordes, etc.

Se usa mucho para recuperar el hdc de Windows o de lo que hay en pantalla en ese momento.


Ejemplo


Hdc = GetWindowDC(frmName.hwnd)


O se quieres recuperar el hdc de la pantalla, es decir, de windows pones:


GetWindowDC(0).

Ejm. Copien esto en un formulario pongan el formulario en la esquina superior izquierda. Y luego presionan un botón, el cual tienen que incluir en el formulario.

Private Declare Function GetWindowDC Lib "user32" (ByVal hwnd As Long) As Long

Private Declare Function TextOut Lib "gdi32" Alias "TextOutA" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long, ByVal lpString As String, ByVal nCount As Long) As Long

Private Sub Command1_Click()

    TextOut GetWindowDC(0), 500, 500, "HOLA", 4

End Sub

Verán que cuando ejecuten el programa y presionen el botón imprimirá “HOLA” afuera del formulario. Esto es por lo que explique el GetWindowDC(0) retorna el DC de toda el área gráfica de Windows.

RELEASEDC

Api: Declare Function ReleaseDC Lib "user32" Alias "ReleaseDC" (ByVal hwnd As Long, ByVal hdc As Long) As Long

Se utiliza para remover un DC o HDC de la memoria. Úsenlo cuando utilicen las API’S CreateDC y CreateCompatibleDC.

Ejm:

del = ReleaseDC(hwnd, hdc)

El API retorna el valor de TRUE se ejecución fue un éxito.

EJEMPLO EN VB

Bueno ya hemos visto algunas API que trabajan con el Dispositivo de Contexto ahora vamos a dibujar y a realizar ejemplos en VB que les aseguro que algunos le parecerán muy interesantes. 

Pare el ejemplo necesitaremos 3 Pictures, en el Picture2 coloca una  imagen con un fondo blanco, en el Picture2 coloca la imagen que quieras y en el Picture3 déjalo sin nada, luego coloca 9 botones y continuación copia el siguiente código:

Private Type BLENDFUNCTION

  BlendOp As Byte

  BlendFlags As Byte

  SourceConstantAlpha As Byte

  AlphaFormat As Byte

End Type

Private Type rect

    Left As Long

    Top As Long

    Right As Long

    Bottom As Long

End Type

Private Declare Function Rectangle Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Private Declare Function Ellipse Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Private Declare Function Arc Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long, ByVal X3 As Long, ByVal Y3 As Long, ByVal X4 As Long, ByVal Y4 As Long) As Long

Private Declare Function DrawText Lib "user32" Alias "DrawTextA" (ByVal hdc As Long, ByVal lpStr As String, ByVal nCount As Long, lpRect As rect, ByVal wFormat As Long) As Long

Private Declare Function BitBlt Lib "gdi32" (ByVal hDestDC As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal dwRop As Long) As Long

Private Declare Function TransparentBlt Lib "msimg32.dll" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal nSrcWidth As Long, ByVal nSrcHeight As Long, ByVal crTransparent As Long) As Boolean

Private Declare Function AlphaBlend Lib "msimg32.dll" (ByVal hdc As Long, ByVal lInt As Long, ByVal lInt As Long, ByVal lInt As Long, ByVal lInt As Long, ByVal hdc As Long, ByVal lInt As Long, ByVal lInt As Long, ByVal lInt As Long, ByVal lInt As Long, ByVal BLENDFUNCT As Long) As Long

Private Declare Function SetPixel Lib "gdi32" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long, ByVal crColor As Long) As Long

Private Declare Function GetPixel Lib "gdi32" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long) As Long

Private Declare Function SetTextColor Lib "gdi32" (ByVal hdc As Long, ByVal crColor As Long) As Long

Private Declare Function CreatePen Lib "gdi32" (ByVal nPenStyle As Long, ByVal nWidth As Long, ByVal crColor As Long) As Long

Private Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Private Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Private Declare Sub RtlMoveMemory Lib "kernel32.dll" (Destination As Any, Source As Any, ByVal Length As Long)

Const PS_SOLID = 0

Const DT_CENTER = &H1

Dim rect As rect

Dim blend As BLENDFUNCTION

Private Sub Command1_Click()

    Picture3.Cls

    Dim penold As Long

    Dim pennew As Long

    pennew = CreatePen(PS_SOLID, 2, RGB(Int(255 * Rnd) + 1, Int(255 * Rnd) + 1, Int(255 * Rnd) + 1))

    penold = SelectObject(Picture3.hdc, pennew)

    Rectangle Picture3.hdc, 0, 0, 100, 100

    SelectObject Picture3.hdc, penold

    DeleteObject pennew

End Sub

Private Sub Command2_Click()

    Picture3.Cls

    Dim penold As Long

    Dim pennew As Long

    pennew = CreatePen(PS_SOLID, 2, RGB(Int(255 * Rnd) + 1, Int(255 * Rnd) + 1, Int(255 * Rnd) + 1))

    penold = SelectObject(Picture3.hdc, pennew)

    Ellipse Picture3.hdc, 0, 0, 100, 100

    SelectObject Picture3.hdc, penold

    DeleteObject pennew

End Sub

Private Sub Command3_Click()

    Picture3.Cls

    Dim penold As Long

    Dim pennew As Long

    pennew = CreatePen(PS_SOLID, 2, RGB(Int(255 * Rnd) + 1, Int(255 * Rnd) + 1, Int(255 * Rnd) + 1))

    penold = SelectObject(Picture3.hdc, pennew)

    Arc Picture3.hdc, 0, 0, 100, 100, 200, 200, 0, 0

    SelectObject Picture3.hdc, penold

    DeleteObject pennew

End Sub

Private Sub Command4_Click()

    Picture3.Cls

    SetTextColor Picture3.hdc, RGB(Int(255 * Rnd) + 1, Int(255 * Rnd) + 1, Int(255 * Rnd) + 1)

    rect.Left = 0

    rect.Top = 0

    rect.Bottom = 200

    rect.Right = 200

    DrawText Picture3.hdc, "HOLA MUNDO", 10, rect, DT_CENTER

End Sub

Private Sub Command5_Click()

    Dim Valor As Long

    Valor = GetPixel(Picture1.hdc, 10, 10)

    MsgBox "El Valor o el color del PIXEL 10X10 es: " & Hex(Valor)

End Sub

Private Sub Command6_Click()

    Dim Valor As Long

    Valor = GetPixel(Picture1.hdc, 10, 10)

    MsgBox "El Valor o el color del PIXEL 10X10 es: " & Hex(Valor)

    SetPixel Picture1.hdc, 10, 10, &H0

    Valor = GetPixel(Picture1.hdc, 10, 10)

    MsgBox "Ahora el Valor o el color del PIXEL 10X10 cambia a: " & Hex(Valor)

End Sub

Private Sub Command7_Click()

    BitBlt Picture2.hdc, 0, 0, 100, 100, Picture1.hdc, 0, 0, vbSrcCopy

End Sub

Private Sub Command8_Click()

    TransparentBlt Picture1.hdc, 0, 0, 100, 100, Picture2.hdc, 0, 0, 100, 100, vbWhite

End Sub

Private Sub Command9_Click()

    Dim vlong As Long

    With blend

        .BlendOp = AC_SRC_OVER

        .BlendFlags = 0

        .SourceConstantAlpha = 80

        .AlphaFormat = 0

    End With

    RtlMoveMemory vlong, blend, 4

    AlphaBlend Picture1.hdc, 0, 0, 100, 100, Picture2.hdc, 0, 0, 100, 100, vlong

End Sub

Private Sub Form_Load()

Command1.Caption = "Rectangulo"

Command2.Caption = "Elipse"

Command3.Caption = "Arc"

Command4.Caption = "Texto"

Command5.Caption = "GetPixel"

Command6.Caption = "SetPixel"

Command7.Caption = "BitBlt"

Command8.Caption = "TransparentBlt"

Command9.Caption = "AlphaBlend"

End Sub

Si ejecutan el ejercicio y empiezan a presionar los botones van a ver cosas muy interesantes, que a continuación voy a explicar.

Command1

Private Sub Command1_Click()

    Picture3.Cls

    Dim penold As Long

    Dim pennew As Long

    pennew = CreatePen(PS_SOLID, 2, RGB(Int(255 * Rnd) + 1, Int(255 * Rnd) + 1, Int(255 * Rnd) + 1))    

    penold = SelectObject(Picture3.hdc, pennew)

    Rectangle Picture3.hdc, 0, 0, 100, 100

    SelectObject Picture3.hdc, penold

    DeleteObject pennew

End Sub

RECTANGLE


Api: Private Declare Function Rectangle Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Considero que no hay problema con esta API, tan sencillamente dibuja un rectángulo pasándole los cuatros parámetros necesarios para dibujarlo, X1 (Left), Y1(Top), X2 (Right), Y2(Bottom).

Las demás API presentadas en el botón Command1 ya las vimos en el inicio de las API’S Gráficas.

Command2

Private Sub Command2_Click()

    Picture3.Cls

    Dim penold As Long

    Dim pennew As Long

    pennew = CreatePen(PS_SOLID, 2, RGB(Int(255 * Rnd) + 1, Int(255 * Rnd) + 1, Int(255 * Rnd) + 1))

    penold = SelectObject(Picture3.hdc, pennew)

    Ellipse Picture3.hdc, 0, 0, 100, 100

    SelectObject Picture3.hdc, penold

    DeleteObject pennew

End Sub

ELLIPSE

Api: Private Declare Function Ellipse Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Prácticamente igual a Rectangle, con la única diferencia que dibuja una Ellipse.

Command3

Private Sub Command3_Click()

    Picture3.Cls

    Dim penold As Long

    Dim pennew As Long

    pennew = CreatePen(PS_SOLID, 2, RGB(Int(255 * Rnd) + 1, Int(255 * Rnd) + 1, Int(255 * Rnd) + 1))

    penold = SelectObject(Picture3.hdc, pennew)

    Arc Picture3.hdc, 0, 0, 100, 100, 200, 200, 0, 0

    SelectObject Picture3.hdc, penold

    DeleteObject pennew

End Sub

ARC

Api: Private Declare Function Arc Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long, ByVal X3 As Long, ByVal Y3 As Long, ByVal X4 As Long, ByVal Y4 As Long) As Long
Esta API es utilizada para dibujar ARCOS en un HDC, los parámetros X1,X2,Y1 y Y2, determinan el rectángulo en donde se dibujara el arco, y los parámetros X3,X4,Y3 y Y4 determinan la dimensión del arco. Intente cambiar los valores de los últimos 4 parámetros y vean lo que sucede.

Command4

Private Sub Command4_Click()

    Picture3.Cls

    SetTextColor Picture3.hdc, RGB(Int(255 * Rnd) + 1, Int(255 * Rnd) + 1, Int(255 * Rnd) + 1)

    rect.Left = 0

    rect.Top = 0

    rect.Bottom = 200

    rect.Right = 200

    DrawText Picture3.hdc, "HOLA MUNDO", 10, rect, DT_CENTER | DT_SINGLELINE

End Sub

DRAWTEXT

Api: Private Declare Function DrawText Lib "user32" Alias "DrawTextA" (ByVal hdc As Long, ByVal lpStr As String, ByVal nCount As Long, lpRect As rect, ByVal wFormat As Long) As Long
Esta API dibuja una cadena de texto en un punto especifico del HDC, entre sus parámetros tenemos a un string “Hola Mundo” que es la cadena de texto que queremos mostrar, tenemos un número “10”, que es la longitud de la cadena, un “rect”, que es el rectángulo donde dibujamos la cadena y wFormat que le damos el valor de una constante DT_CENTER y DT_SINGLELINE, noten que si quiero usar mas de un valor lo separo con el operador logico OR ( | ).


Valores mas usados en wFormat:


Public Const DT_CENTER = &H1

Public Const DT_BOTTOM = &H8

Public Const DT_LEFT = &H0

Public Const DT_MULTILINE = ( &H00000001)

Public Const DT_RIGHT = &H2

Public Const DT_VCENTER = &H4

Public Const DT_SINGLELINE = &H20

.

Etc.

Command5

Private Sub Command5_Click()

    Dim Valor As Long

    Valor = GetPixel(Picture1.hdc, 10, 10)    

    MsgBox "El Valor o el color del PIXEL 10X10 es: " & Hex(Valor)

End Sub

GETPIXEL

Api: Private Declare Function GetPixel Lib "gdi32" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long) As Long

Esta API retorna el valor o podemos decir que retorna el color (Recuerden que en la computadora los colores se representan con números), de un PIXEL específico.

Sus parámetros son, el HDC, X, Y. Pueden notar que en esta API si nos interesa el valor de retorno. Luego el Valor lo convierto en Hexadecimal con la función Hex de VB. Y muestro su valor a través de un mensaje. Esta API es buena para hacer comparación de imágenes.

Command6

Private Sub Command6_Click()

    Dim Valor As Long

    Valor = GetPixel(Picture1.hdc, 10, 10)    

    MsgBox "El Valor o el color del PIXEL 10X10 es: " & Hex(Valor)    

    SetPixel Picture1.hdc, 10, 10, &H0    

    Valor = GetPixel(Picture1.hdc, 10, 10)    

    MsgBox "Ahora el Valor o el color del PIXEL 10X10 cambia a: " & Hex(Valor)

End Sub

SETPIXEL

Api: Private Declare Function SetPixel Lib "gdi32" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long, ByVal crColor As Long) As Long

Con esta API le asignamos el color a un PÍXEL especifico, sus parámetros son iguales a GetPixel con la diferencia que su último parámetro es el color el cual queremos asignar a ese píxel. 

Lo que hago en el ejemplo es mostrar el píxel actual, luego lo pinto de color negro, y vuelva a usar GetPixel para que vean que si cambia el color del píxel.

Command7

Private Sub Command7_Click()

    BitBlt Picture2.hdc, 0, 0, 100, 100, Picture1.hdc, 0, 0, vbSrcCopy

End Sub

BITBLT

Api: Private Declare Function BitBlt Lib "gdi32" (ByVal hDestDC As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal dwRop As Long) As Long

Esta API es muy usado por los programadores ella permite copiar una imagen completa o una porción de la misma, en otra imagen.

Parámetros:

hDestDC = Es el HDC destino donde se copiará la imagen.

x,y = son las coordenadas en donde se empezará a dibujar la imagen.

nWidth,nHeight = son las medidas para establecer el tamaño de lo que vamos a copiar.

hSrcDC = es el  HDC de donde vamos a cortar la imagen, es decir, es el origen.

xSrc, ySrc = son las coordenadas en donde se empezará a cortar la imagen en el origen.

dwRop = es la forma del como se hace la copia. En realidad yo solo he usado el vbScroCopy pero existen otras constantes que se las presento a continuación.

Public Const BLACKNESS = &H42 ' (DWORD) dest = BLACK

Public Const DSTINVERT = &H550009    ' (DWORD) dest = (NOT dest)

Public Const MERGECOPY = &HC000CA    ' (DWORD) dest = (source AND pattern)

Public Const MERGEPAINT = &HBB0226   ' (DWORD) dest = (NOT source) OR dest

Public Const NOTSRCCOPY = &H330008   ' (DWORD) dest = (NOT source)

Public Const NOTSRCERASE = &H1100A6   ' (DWORD) dest = (NOT src) AND (NOT dest)

Public Const PATCOPY = &HF00021 ' (DWORD) dest = pattern

Public Const PATINVERT = &H5A0049    ' (DWORD) dest = pattern XOR dest

Public Const PATPAINT = &HFB0A09    ' (DWORD) dest = DPSnoo

Public Const SRCAND = &H8800C6 ' (DWORD) dest = source AND dest

Public Const SRCCOPY = &HCC0020 ' (DWORD) dest = source

Public Const SRCERASE = &H440328    ' (DWORD) dest = source AND (NOT dest )

Public Const SRCINVERT = &H660046    ' (DWORD) dest = source XOR dest

Public Const SRCPAINT = &HEE0086    ' (DWORD) dest = source OR dest

Public Const WHITENESS = &HFF0062    ' (DWORD) dest = WHITE

El tamaño de la imagen esta determinado por los parámetros nWidth y nHeight.

Con esta API ustedes pueden diseñar lo que muchos tratan de buscar cuando manejan imágenes en VB, todo sabemos que cuando una imagen es mas grande que el control Picture donde esta contenida, este control no activa un Scroll Automático. Pero colocando el control Scroll junto con el control Picture, ustedes pueden programar que si la imagen es mas grande que el Picture al darle click al control Scroll en cualquiera de sus flechas les mostrará lo que sigue de la imagen. Esta parte se les dejo de tarea. Lo que les puedo decir  que difícil NO ES!!.

Command8

Private Sub Command8_Click()

    TransparentBlt Picture1.hdc, 0, 0, 100, 100, Picture2.hdc, 0, 0, 100, 100, vbWhite

End Sub

TRANSPARENTBLT

Api: Private Declare Function TransparentBlt Lib "msimg32.dll" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long, ByVal nWidth As Long, ByVal nHeight As Long, ByVal hSrcDC As Long, ByVal xSrc As Long, ByVal ySrc As Long, ByVal nSrcWidth As Long, ByVal nSrcHeight As Long, ByVal crTransparent As Long) As Boolean

Esta API es muy importante, los primeros 10 parámetros ya deben de saber que hacen, que es similar a las API de arriba, lo mas importante es el parámetro crTransparent. Si me hicieron caso y pusieron una imagen de fondo blanco en el Picture2. Notarán que la imagen se copia en el Picture1 pero sin el fondo blanco. Eso se debe ya que colocamos en el parámetro crTransparent el valor de vbWhite. Si su imagen hubiese usado azul tendrían que cambiar el vbWhite por vbBlue.

En pocas palabras esta API hace los mismo que hacen con las películas. Vemos en las películas que cuando quieren hacer interactuar una imagen REAL con una de computadora, graban la real donde el personaje se pone en un escenario AZUL luego la máquina quita el color Azul y pone los píxeles de colores que tiene la imagen de la computadora. Esta API hace eso pero en ves, de con videos lo hace con Imágenes.

Command9

Private Sub Command9_Click()

    Dim vlong As Long

    With blend

        .BlendOp = AC_SRC_OVER

        .BlendFlags = 0

        .SourceConstantAlpha = 80

        .AlphaFormat = 0

    End With

    RtlMoveMemory vlong, blend, 4    

    AlphaBlend Picture1.hdc, 0, 0, 100, 100, Picture2.hdc, 0, 0, 100, 100, vlong

End Sub

Una de las novedades que impresiono en Windows 2000 era que se podían hacer formulario y objetos transparentes. Eso le daba a los formularios un toque de elegancia que a mi parecer es excelente.

Pero Microsoft a los programadores de Windows 98 y ME nos dejo con las ganas de programar los formularios transparentes. Pero introdujo un API que vamos a ver a continuación que permite hacer transparente una imagen encima de otra. 

Si corrieron el ejemplo y presionaron Command9, notaron que una porción de la imagen del picture2 se copio en el picture1 pero con un nivel de transparencia pues esa es el API AlphaBlend que explicaremos a continuación. Es importante mencionar que esta API solamente funciona en Windows 98, ME, NT y 2000.

ALPHABLEND

Api: Declare Function AlphaBlend Lib "msimg32.dll" (ByVal hdc As Long, ByVal lInt As Long, ByVal lInt As Long, ByVal lInt As Long, ByVal lInt As Long, ByVal hdc As Long, ByVal lInt As Long, ByVal lInt As Long, ByVal lInt As Long, ByVal lInt As Long, ByVal BLENDFUNCT As Long) As Long

Esta API sirve para realizar ese efecto tan impresionante como lo es la Transparencia. Y vamos a explicar sus parámetros colocando la línea de código que inserte en el proyecto.

    AlphaBlend Picture1.hdc, 0, 0, 100, 100, Picture2.hdc, 0, 0, 100, 100, vlong

Pueden ver que el primer parámetro es el HDC receptor, es decir, en donde se va a copiar la imagen, los cuatros parámetros siguientes, son los valores de X,Y,WIDTH,HEIGHT, respectivamente, luego tenemos al HDC emisor, es decir, de donde sacamos la imagen, y al igual que el HDC receptor, los cuatros parámetros siguiente son X,Y,WIDTH y HEIGHT. Cambien los valores para que vea que sucede cuando lo hacen.

Tenemos el parámetro número 11. Según la función API es “ByVal BLENDFUNCT As Long”. Aquí notamos algo extraño este parámetro es una estructura pero se pasa como un parámetro de tipo “long” ¿Por qué?, ni idea!.

Vemos que declaramos la estructura y le dimos valores:

    Dim vlong As Long

    With blend

        .BlendOp = AC_SRC_OVER

        .BlendFlags = 0

        .SourceConstantAlpha = 80

        .AlphaFormat = 0

    End With

En donde vi el ejemplo, no vi el uso de otras constantes en BlendOp, solamente vi el AC_SRC_OVER y no explica con detalle el uso de los demás parámetros de la estructura (lei en un libro que todavía los parámetros de BlendOp y SourceConstantAlpha en este momento acepta esos valores nada mas) , pero si estoy claro que el atributo o propiedad que define que transparente va hacer la imagen de origen con respecto  a la imagen destino es “SourceConstantAlpha”, mientras el valor vaya descendiendo la imagen se hará mas transparente, el valor máximo que puede tener es 255. Que es donde la imagen se ve completamente sólida, sin ver ningún tipo de Transparencia.

Pero lo curioso no es la estructura, es esta función API que le sigue a continuación:

    RtlMoveMemory vlong, blend, 4    

Esta función API la ayuda de MSND no la tiene, así que no la va explicar tal cual como es, sino que voy hacer una interpretación lógica de la función.

Ya vimos que la función API AlphaBlend no acepta la estructura como parámetro sino un valor de tipo “long” (Estoy hablando del parámetro número 11), lo que yo entiendo que hace esta API es como convertir la estructura, a una variable de tipo LONG (NOTA: estoy suponiendo que la función API realiza eso, ya que no encontré en la ayuda de MSDN la definición de la misma), ya que si vemos el API en su declaración:

Private Declare Sub RtlMoveMemory Lib "kernel32.dll" (Destination As Any, Source As Any, ByVal Length As Long)


Destination = es “vLogn” variable que declaramos de tipo Long


Source = la estructura blend


Lenght = 4, esto se debe a que la variables de tipo Long son de 4 Bytes.

Luego que convertimos la estructura a una variable de tipo long se la pasamos sin problema a la función API AlphaBlend.

Con todo esto hemos concluido la parte GRÁFICA, cabe decir de que existen muchas mas API’S gráficas y muchas formas de relacionarlas para hacer efectos extraordinarios. Pero yo les doy la iniciativa y ustedes se encargan del resto. Ya si desean profundizar necesitarán comprar o buscar un libro que se dedique a la programacion bajo Windows.

FONTS

Quien haya programado en DOS podrá recordar que cuando uno quería mostrar una línea debajo de otra, lo que hacia era cambiar el valor de la fila y listo!. En Windows lo cosa cambia, en primer lugar tenemos muchos tipos de letras y en segundo lugar no vasta con aumentar en 1 al eje de las “y” para añadir otra línea. A continuación voy a explicar estos dos detalles con un ejemplo que voy a poner a continuación (Para este ejemplo es necesario que pongan un COMBO BOX en el formulario):

Const FW_NORMAL = 400

Const DEFAULT_CHARSET = 1

Const PROOF_QUALITY = 2

Const OPAQUE = 2

Const TRANSPARENT = 1

Const LOGPIXELSY = 90

Private Declare Function DeleteObject Lib "gdi32" (ByVal hObject As Long) As Long

Private Declare Function CreateFont Lib "gdi32" Alias "CreateFontA" (ByVal nHeight As Long, ByVal nWidth As Long, ByVal nEscapement As Long, ByVal nOrientation As Long, ByVal fnWeight As Long, ByVal fdwItalic As Boolean, ByVal fdwUnderline As Boolean, ByVal fdwStrikeOut As Boolean, ByVal fdwCharSet As Long, ByVal fdwOutputPrecision As Long, ByVal fdwClipPrecision As Long, ByVal fdwQuality As Long, ByVal fdwPitchAndFamily As Long, ByVal lpszFace As String) As Long

Private Declare Function SelectObject Lib "gdi32" (ByVal hdc As Long, ByVal hObject As Long) As Long

Private Declare Function TextOut Lib "gdi32" Alias "TextOutA" (ByVal hdc As Long, ByVal x As Long, ByVal y As Long, ByVal lpString As String, ByVal nCount As Long) As Long

Private Declare Function SetBkMode Lib "gdi32" (ByVal hdc As Long, ByVal nBkMode As Long) As Long

Private Declare Function GetDeviceCaps Lib "gdi32" (ByVal hdc As Long, ByVal nIndex As Long) As Long

Private Declare Function MulDiv Lib "kernel32" (ByVal nNumber As Long, ByVal nNumerator As Long, ByVal nDenominator As Long) As Long

Private Declare Function SetBkColor Lib "gdi32" (ByVal hdc As Long, ByVal crColor As Long) As Long

Private Declare Function SetTextColor Lib "gdi32" (ByVal hdc As Long, ByVal crColor As Long) As Long

Private Declare Function GetTextMetrics Lib "gdi32" Alias "GetTextMetricsA" (ByVal hdc As Long, lpMetrics As TEXTMETRIC) As Long

Private Type TEXTMETRIC

    tmHeight As Long

    tmAscent As Long

    tmDescent As Long

    tmInternalLeading As Long

    tmExternalLeading As Long

    tmAveCharWidth As Long

    tmMaxCharWidth As Long

    tmWeight As Long

    tmOverhang As Long

    tmDigitizedAspectX As Long

    tmDigitizedAspectY As Long

    tmFirstChar As Byte

    tmLastChar As Byte

    tmDefaultChar As Byte

    tmBreakChar As Byte

    tmItalic As Byte

    tmUnderlined As Byte

    tmStruckOut As Byte

    tmPitchAndFamily As Byte

    tmCharSet As Byte

End Type

Dim TEXTM As TEXTMETRIC

Dim F, res As Long

Private Sub Combo1_Click()

    SelectObject Me.hdc, res

    DeleteObject F

    F = CreateFont(-MulDiv(10, GetDeviceCaps(Me.hdc, LOGPIXELSY), 72), 0, 0, 0, FW_NORMAL, 0, 0, 0, DEFAULT_CHARSET, OUT_DEFAULT_PRECIS, CLIP_DEFAULT_PRECIS, PROOF_QUALITY, DEFAULT_PITCH, Combo1.Text)

    res = SelectObject(Me.hdc, F)

    GetTextMetrics Me.hdc, TEXTM

    Me.Refresh

End Sub

Private Sub Form_Load()

    Combo1.AddItem "Times New Roman"

    Combo1.AddItem "Arial"

    Combo1.AddItem "MS Serif"

    Combo1.AddItem "Courier"

    Combo1.AddItem "Comic Sans MS"

    Combo1.ListIndex = 0

    Randomize

    F = CreateFont(-MulDiv(10, GetDeviceCaps(Me.hdc, LOGPIXELSY), 72), 0, 0, 0, FW_NORMAL, 0, 0, 0, DEFAULT_CHARSET, OUT_DEFAULT_PRECIS, CLIP_DEFAULT_PRECIS, PROOF_QUALITY, DEFAULT_PITCH, Combo1.Text)

    res = SelectObject(Me.hdc, F)

    GetTextMetrics Me.hdc, TEXTM

End Sub

Private Sub Form_Paint()

    Dim i As Long

    For i = 1 To 10

        SetBkMode Me.hdc, OPAQUE

        SetBkColor Me.hdc, RGB(Int(255 * Rnd) + 1, Int(255 * Rnd) + 1, Int(255 * Rnd) + 1)

        SetTextColor Me.hdc, RGB(Int(255 * Rnd) + 1, Int(255 * Rnd) + 1, Int(255 * Rnd) + 1)

        TextOut Me.hdc, 0, (i * TEXTM.tmHeight), "HOLA MUNDO", Len("HOLA MUNDO")

    Next

    SelectObject Me.hdc, res

    DeleteObject F

End Sub

Al correrlo, verán que 10 líneas alineadas se colocan una debajo de otra, las cuales tienen un color de fondo, y otro color para la letra, a su vez notarán que al cambiar las letras en al combo, cambia en el formulario manteniendo la alineación; a continuación explicamos las API’S usadas:

CREATEFONT
Api: Private Declare Function CreateFont Lib "gdi32" Alias "CreateFontA" (ByVal nHeight As Long, ByVal nWidth As Long, ByVal nEscapement As Long, ByVal nOrientation As Long, ByVal fnWeight As Long, ByVal fdwItalic As Boolean, ByVal fdwUnderline As Boolean, ByVal fdwStrikeOut As Boolean, ByVal fdwCharSet As Long, ByVal fdwOutputPrecision As Long, ByVal fdwClipPrecision As Long, ByVal fdwQuality As Long, ByVal fdwPitchAndFamily As Long, ByVal lpszFace As String) As Long


Esta API se utiliza para cambiar el tipo de letra del HDC, esta API parece muy compleja debido a la cantidad de parámetros que ella posee, pero les aseguro que no son gran cosa, el único parámetro que no voy a explicar va ser nHeight, y ya les diré el por que.


Parámetros

· nHeight = Parámetro que especifica el Alto de la letra, para que el alto sea igual a los tamaños que poseen la letra o fuente hay que usar una formula que es la siguiente:

-MulDiv(10, GetDeviceCaps(Me.hdc, LOGPIXELSY), 72)
Por que colocar esta formula y no colocar el número 10 directo, bueno hagan la prueba, quiten la formula y pongan el número 10 solamente, comparen el tamaño de la letra, con la misma letra vista desde WORD, verán de que es mas pequeña, que la letra de WORD, el ¿por que de esto?, no lo se, es decir, no se por que el uso de la formula, la formula la usan para que cuando ustedes pongan el número del size, en donde esta el 10, el tamaño sea igual a lo que vemos en otro programas. Ahora cambien el 10 que esta después del paréntesis -MulDiv(10 – y pónganle 16 por ejemplo - MulDiv(16 – al ejecutarlo, pongan en WORD un texto con la misma letra y de tamaño 16 y verán que son del mismo tamaño, pero si quitan la formula y ponen 16, verán de que no es igual. Así que les debo una explicación, pero al igual que ustedes yo estoy aprendiendo y me falta mucho por aprender.

· nWidth = determina el ancho de la letra. Aumenten el número y vean lo que sucede.

· nEscapement = Pongan 180 y entenderán que hace.

· fnWeight = determina al ancho de la letra. Valores que puede tomar:

Const FW_DONTCARE = 0

Const FW_THIN = 100

Const FW_EXTRALIGHT = 200

Const FW_LIGHT = 300

Const FW_NORMAL = 400

Const FW_MEDIUM = 500

Const FW_SEMIBOLD = 600

Const FW_BOLD = 700

Const FW_EXTRABOLD = 800

Const FW_HEAVY = 900

Const FW_BLACK = FW_HEAVY

Const FW_DEMIBOLD = FW_SEMIBOLD

Const FW_REGULAR = FW_NORMAL

Const FW_ULTRABOLD = FW_EXTRABOLD

Const FW_ULTRALIGHT = FW_EXTRALIGHT

· fdwItalic = Toma como valores TRUE o FALSE y determina si la letra va de forma cursiva o no.

· fdwUnderline = Toma como valores TRUE o FALSE y determina si la letra va subrayada.

· fdwStrikeOut =  Toma como valores TRUE o FALSE y determina si la letra va Tachada

· fdwCharSet = Especifica el tipo de carácter. Valores que puede tomar:

Const ANSI_CHARSET = 0

Const DEFAULT_CHARSET = 1

Const SYMBOL_CHARSET = 2

Const SHIFTJIS_CHARSET = 128

Const HANGEUL_CHARSET = 129

Const CHINESEBIG5_CHARSET = 136

Const OEM_CHARSET = 255

· fdwOutputPrecision = Valores que puede tomar:

            Const OUT_CHARACTER_PRECIS = 2

Const OUT_DEFAULT_PRECIS = 0

Const OUT_DEVICE_PRECIS = 5

· fdwClipPrecision = Valores que puede tomar:

Const CLIP_DEFAULT_PRECIS = 0

Const CLIP_CHARACTER_PRECIS = 1

Const CLIP_STROKE_PRECIS = 2

· fdwQuality = Valores que puede tomar:

Const DEFAULT_QUALITY = 0

Const DRAFT_QUALITY = 1

Const PROOF_QUALITY = 2

· fdwPitchAndFamily = Valores que puede tomar:

Const DEFAULT_PITCH = 0

Const FIXED_PITCH = 1

Const VARIABLE_PITCH = 2

· lpszFace = Nombre de la letra.

Hay algunos parámetro de los cuales no muestro explicación, en realidad es por que nunca los he usado, ahí les pase las constantes que pueden usar para cada parámetro y les dejo a ustedes la prueba de ello, normalmente se usa esta API para cambiar el tipo de letra del HDC, y usualmente estos valores no se cambian.

SETBKMODE
Api: Private Declare Function SetBkMode Lib "gdi32" (ByVal hdc As Long, ByVal nBkMode As Long) As Long

Esta API determina si el fondo o el rectángulo en donde se dibuja la letra o fuente es opaco o transparente. El API en el ejemplo lo usamos de esta manera:

        SetBkMode Me.hdc, OPAQUE

Es por esa razón que en donde se escribe HOLA MUNDO, hay un fondo de color, si en ves de colocar OPAQUE colocan TRANSPARENT el fondo seria el del formulario que lo contiene, esto es igual a la propiedad “BackStyle” del control LABEL de VB.

Los parámetros son el HDC, y el valor del BkMode que puede ser:

Const OPAQUE = 2

Const TRANSPARENT = 1

SETBKCOLOR

Api: Private Declare Function SetBkColor Lib "gdi32" (ByVal hdc As Long, ByVal crColor As Long) As Long

Cuando SetBkMode esta en OPAQUE el color que muestra de fondo es el color que haya sido asignado con esta API.

SetBkColor Me.hdc, RGB(Int(255 * Rnd) + 1, Int(255 * Rnd) + 1, Int(255 * Rnd) + 1)

En el ejemplo para cada línea que dibujamos elige un color al azar. Creo que sus parámetros hablan por si solo.

SETTEXTCOLOR

Api: Private Declare Function SetTextColor Lib "gdi32" (ByVal hdc As Long, ByVal crColor As Long) As Long


Determina el color de la letra. Tiene dos parámetros el HDC y el Color.

GETTEXTMETRICS

Api: Private Declare Function GetTextMetrics Lib "gdi32" Alias "GetTextMetricsA" (ByVal hdc As Long, lpMetrics As TEXTMETRIC) As Long

Llegamos a un API muy importante, se recuerdan lo que dije al inicio del tema FONT, que en DOS cuando ustedes querían mostrar el texto en otra línea lo que hacia era aumentar o disminuir el eje de las Y, bueno pues Windows no es así, podrán notar que para dibujar las 10 líneas, no puse 10 veces TEXTOUT, sino que lo introduje en un ciclo FOR. También notaran que las líneas están alineadas una debajo de otras. Ahora ustedes se preguntan ¿CÓMO LO HICE?

Bueno la respuestas no están complicada, en Windows existe una estructura que se llama TEXTMETRICS. La cual muestro a continuación:

Private Type TEXTMETRIC

    tmHeight As Long  / Altura de la letra o fuente

    tmAscent As Long 

    tmDescent As Long

    tmInternalLeading As Long

    tmExternalLeading As Long

    tmAveCharWidth As Long / Anchura media

    tmMaxCharWidth As Long / Anchura Maxima
    tmWeight As Long / Grosor o Peso de la letra o fuente

    tmOverhang As Long

    tmDigitizedAspectX As Long

    tmDigitizedAspectY As Long

    tmFirstChar As Byte / Primer carácter de la fuente
    tmLastChar As Byte / Último carácter de la fuente

    tmDefaultChar As Byte / carácter por defecto

    tmBreakChar As Byte

    tmItalic As Byte / Indica si la letra esta cursiva

    tmUnderlined As Byte / Indica si la letra esta subrayada

    tmStruckOut As Byte / Indica si la letra esta tachada
    tmPitchAndFamily As Byte

    tmCharSet As Byte

End Type

El explicar detalladamente esta estructura significa poner un dibujo, donde se muestre la letra a nivel de píxeles, como no voy a profundizar en este tema, les mostré arriba para que sirven nada mas las propiedades de la estructura de las cuales yo considero mas importante. En los libros de Programación de Windows encontraran una explicación detallada de cada una de las propiedades de la estructura.

La propiedad que yo use para el programa fue tmHeight, como en Windows se dibuja a nivel de PÍXEL, y tomando en cuenta que cada tipo de letra tiene un tamaño distinto, es decir, no es lo mismo el tamaño del ARIAL que la del TIME NEW ROMAN, ni tampoco es igual la altura de la letra “o”, con el alto de la letra “i”, tmHeight me retorna el alto máximo de la fuente seleccionada, eso lo hice en:

    F = CreateFont(-MulDiv(10, GetDeviceCaps(Me.hdc, LOGPIXELSY), 72), 0, 0, 0, FW_NORMAL, False, False, False, DEFAULT_CHARSET, 2, CLIP_DEFAULT_PRECIS, PROOF_QUALITY, DEFAULT_PITCH, Combo1.Text)

    res = SelectObject(Me.hdc, F)

    GetTextMetrics Me.hdc, TEXTM

Primero use CreateFont para crear la letra o fuente, luego se lo asigne al HDC con SelectObject, y luego cuando el HDC tenia configurada la fuente use el API GetTextMetrcis, que lo que hace es llenar la estructura TEXTMETRICS, en este caso la variable “TEXTM”, con la información de la fuente o letra. Para luego usar el tmHeight en PAINT:

For i = 1 To 10

SetBkMode Me.hdc, 1

SetBkColor Me.hdc, RGB(Int(255 * Rnd) + 1, Int(255 * Rnd) + 1, Int(255 * Rnd) + 1)

SetTextColor Me.hdc, RGB(Int(255 * Rnd) + 1, Int(255 * Rnd) + 1, Int(255 * Rnd) + 1)

TextOut Me.hdc, 0, (i * TEXTM.tmHeight), "HOLA MUNDO", Len("HOLA MUNDO")

Next

Aquí multiplico el número de “línea”, por el tmHeight, para lograr sincronizar y dar un aspecto elegante a la presentación de las líneas, y podrán notar que no importaba que fuente o letra eligiesen pero siempre se muestra de forma ordenada.

REGISTRO DE WINDOWS

Antes de empezar a explicar realicen estos pasos:

1. Dale al botón “Inicio” de Windows.

2. Y dale click en el enlace EJECUTAR.

3. Cuando veas la ventana de “Ejecutar” coloca en la Caja de Texto que te aparece: el comando “regedit”, y luego presiona “Aceptar”

Cuando se abra, ya estarás adentro del registro de Windows. El registro de Windows, es donde están todas las configuraciones tanto de Windows como de los programas instalados en él, así como los registros de cada dll, o control usado por él, es decir, en pocas palabras en el registro se encuentran todas las configuraciones del sistema operativo. Por esa razón, esta bien que veas y te metas por el registro para ver lo que tiene, pero NO BORRES! Ninguna clave o valor que desconozcas.

Lo primero que ves, cuando ejecutas el programa son 6 carpetas HKEY_CLASSES_ROOT, HKEY_CURRENT_USER, HKEY_LOCAL_MACHINE, HKEY_USERS, HKEY_CURRENT_CONFIG, HKEY_DYN_DATA. Cada una de estas carpetas o claves no puede ser borradas y no se pueden añadir mas. Siempre cuando queramos ingresar una nueva carpeta será dentro de una de ellas. 

Ahora ¿cuando o para que se usa el registro?. Cuando un programa necesita guardar valores de inicialización de nuestra aplicación es recomendable dos cosas: 1. Usar archivos .ini, o 2. Usar el registro. Personalmente yo uso el registro, por que tiene las siguientes ventajas con respecto a los archivos .ini: 1. Es fácil que un usuario borre un archivo .ini que una carpeta en el registro. 2. Windows hace respaldo del registro y por ende respalda nuestra información. Ahora yo uso los archivos .ini cuando la información de inicialización de mi aplicación es muy grande. 

Llamo “información de inicialización de mi aplicación”, a los datos o valores que necesita mi programa para arrancar. Obviamente esto dependerá de cada programador y de su forma de diseñar los programas.

A continuación muestro un ejemplo sencillo de las API’S que manejan el registro de Windows, para este ejemplo es necesario un formulario con 5 botones.

Const REG_SZ = 1 

Const REG_BINARY = 3 

Const HKEY_CURRENT_USER = &H80000001

Const ERROR_SUCCESS = 0&

Private Declare Function RegCloseKey Lib "advapi32.dll" (ByVal hKey As Long) As Long

Private Declare Function RegCreateKey Lib "advapi32.dll" Alias "RegCreateKeyA" (ByVal hKey As Long, ByVal lpSubKey As String, phkResult As Long) As Long

Private Declare Function RegDeleteValue Lib "advapi32.dll" Alias "RegDeleteValueA" (ByVal hKey As Long, ByVal lpValueName As String) As Long

Private Declare Function RegOpenKey Lib "advapi32.dll" Alias "RegOpenKeyA" (ByVal hKey As Long, ByVal lpSubKey As String, phkResult As Long) As Long

Private Declare Function RegQueryValueEx Lib "advapi32.dll" Alias "RegQueryValueExA" (ByVal hKey As Long, ByVal lpValueName As String, ByVal lpReserved As Long, lpType As Long, lpData As Any, lpcbData As Long) As Long

Private Declare Function RegSetValueEx Lib "advapi32.dll" Alias "RegSetValueExA" (ByVal hKey As Long, ByVal lpValueName As String, ByVal Reserved As Long, ByVal dwType As Long, lpData As Any, ByVal cbData As Long) As Long

Private Declare Function RegDeleteKey Lib "advapi32.dll" Alias "RegDeleteKeyA" (ByVal hKey As Long, ByVal lpSubKey As String) As Long

Private Sub Command1_Click()

Dim res As Long

resultado = RegCreateKey(HKEY_CURRENT_USER, "PruebaAPI", res)

If resultado = ERROR_SUCCESS Then

    RegCloseKey res

Else

    MsgBox "No se pudo crear la clave"

End If

End Sub

Private Sub Command2_Click()

Dim res As Long

RegOpenKey HKEY_CURRENT_USER, "PruebaAPI", res

RegSetValueEx res, "Valor String", 0, REG_SZ, ByVal "HOLA", Len("HOLA")

RegSetValueEx res, "Valor Binario", 0, REG_BINARY, 10, 4

RegCloseKey res

End Sub

Private Sub Command3_Click()

Dim cadena As String

Dim binario As Long

cadena = String(255, Chr(0))

Dim res As Long

RegOpenKey HKEY_CURRENT_USER, "PruebaAPI", res

RegQueryValueEx res, "Valor String", 0, REG_SZ, ByVal cadena, Len(cadena)

RegQueryValueEx res, "Valor Binario", 0, REG_BINARY, binario, 4

MsgBox cadena

MsgBox binario

RegCloseKey res

End Sub

Private Sub Command4_Click()

Dim res As Long

RegOpenKey HKEY_CURRENT_USER, "PruebaAPI", res

RegDeleteValue res, "Valor String"

RegCloseKey res

End Sub

Private Sub Command5_Click()

RegDeleteKey HKEY_CURRENT_USER, "PruebaAPI"

End Sub

Private Sub Form_Load()

Command1.Caption = "NUEVO CLAVE (KEY)"

Command2.Caption = "NUEVO VALOR"

Command3.Caption = "CONSULTAR VALOR"

Command4.Caption = "BORRAR VALOR"

Command5.Caption = "BORRAR LLAVE"

End Sub

A continuación explico cada API usada en este ejemplo:

RegCreateKey 

Api: Private Declare Function RegCreateKey Lib "advapi32.dll" Alias "RegCreateKeyA" (ByVal hKey As Long, ByVal lpSubKey As String, phkResult As Long) As Long

Esta API crea una nueva “CLAVE” o “Capeta” adentro de una de las 6 carpetas que mencione al inicio, en este ejemplo usamos HKEY_CURRENT_USER.

Parámetros:

hKey = indica adonde almacenamos la nueva clave.

Valores que puede tomar:

Const HKEY_LOCAL_MACHINE = &H80000002

Const HKEY_DYN_DATA = &H80000006

Const HKEY_CURRENT_USER = &H80000001

Const HKEY_CURRENT_CONFIG = &H80000005

Const HKEY_CLASSES_ROOT = &H80000000

Const HKEY_USERS = &H80000003


lpSubKey = Es el string o el nombre que le vamos a dar a la carpeta.

phkResult = es el Handler de la nueva carpeta.

RegCloseKey 

Api: Private Declare Function RegCloseKey Lib "advapi32.dll" (ByVal hKey As Long) As Long


Esta API cierra un registro que haya sido abierto. Toma como único parámetro el Handler del registro abierto.

Vamos a ver estas dos API con el ejemplo del programa:

Private Sub Command1_Click()

Dim res As Long

resultado = RegCreateKey(HKEY_CURRENT_USER, "PruebaAPI", res)

If resultado = ERROR_SUCCESS Then

    RegCloseKey res

Else

    MsgBox "No se pudo crear la clave"

End If

End Sub

Primero declaramos la variable de tipo LONG (“res”) para almacenar el Handler del nuevo registro. Luego le pasamos a RegCreateKey tres parámetros:

· El primer parámetro es HKEY_CURRENT_USER, esto quiere decir, que es ahí donde almacenaremos nuestra nueva carpeta o clave.

· “PruebaAPI” es el nombre que le doy a nuestra carpeta

· Y “res” recibirá el Handler de la carpeta creada.

“If resultado = ERROR_SUCCESS then”

Si el registro fue creado satisfactoriamente entonces lo cerramos, con:

    “RegCloseKey res”

Si corren el programa y le dan al Primer botón, y luego se van a la ventana del registro, le dan doble click a la carpeta HKEY_CURRENT_USER notaran que esta una nueva carpeta llamada “PruebaAPI”, si no esta la carpeta presionen F5 para refrescar el registro. 

RegOpenKey 

Api: Private Declare Function RegOpenKey Lib "advapi32.dll" Alias "RegOpenKeyA" (ByVal hKey As Long, ByVal lpSubKey As String, phkResult As Long) As Long

Esta API se utiliza para abrir una clave o una carpeta del registro. Abrir el registro es necesario para manipularlo.

Parámetro:

hKey = indica adonde almacenamos la clave. Toma los mismo valores que el hKey de RegCreateKey

lpSubKey = Es la carpeta o clave que esta adentro de hKey.

phkResult =  Es el Handler de la carpeta o clave seleccionada.

RegSetValueEx
Api: Private Declare Function RegSetValueEx Lib "advapi32.dll" Alias "RegSetValueExA" (ByVal hKey As Long, ByVal lpValueName As String, ByVal Reserved As Long, ByVal dwType As Long, lpData As Any, ByVal cbData As Long) As Long

Esta API se utiliza para guardar valores en la carpeta o clave seleccionada.

Parámetros:

hKey = en este caso este parámetro no es igual al de RegOpenKey y RegCreateKey, en este caso va el Handler devuelto en RegOpenKey.

lpValueName = El nombre del valor o vamos llamarlo del campo que va a estar adentro en la carpeta.

Reserved = En este parámetro se pasa 0.

dwType = Indica el Tipo de valor que se introducirá. Valores mas usados:

Const REG_BINARY = 3           

Const REG_DWORD = 4           

Const REG_SZ = 1             

lpData = El dato que se le va a pasar al Valor o Campo.

cbData = Tamaño del dato que estamos pasando.

Vamos a ver estas dos API con el ejemplo del programa:

Private Sub Command2_Click()

Dim res As Long

RegOpenKey HKEY_CURRENT_USER, "PruebaAPI", res

RegSetValueEx res, "Valor String", 0, REG_SZ, ByVal "HOLA", Len("HOLA")

RegSetValueEx res, "Valor Binario", 0, REG_BINARY, 10, 4

RegCloseKey res

End Sub

Primero abrimos la carpeta que ya esta creada con RegOpenKey, creo que al ver la explicación de los parámetros y como se esta usando en el ejemplo es fácil de entender el API.

Luego que ya esta abierta procedemos a guardar los valores. Vamos a dejar en claro una cosa. El “valor” es el campo que tiene la carpeta, es como el nombre de una variable, y el “dato” , es el contenido que tiene el “valor”, se que suena un poco enredado para vamos a explicar el API con el ejemplo:

RegSetValueEx res, "Valor String", 0, REG_SZ, ByVal "HOLA", Len("HOLA")

Como primer parámetro pasamos “res”, que es el Handler devuelto por RegOpenKey, luego como segundo parámetro pasamos “Valor String”, este es el nombre del “valor”, el tercer parámetro es 0, luego especificamos el tipo de dato, en este caso como es String le pasamos en valor de la constante REG_SZ, como quinto parámetro pasamos la cadena string a guardar (esto es el “dato”), y como último parámetro le pasamos la longitud de la cadena.

La misma explicación se vale para el valor binario. Luego después de almacenar nuestros valores procedemos a cerrar el Registro con RegCloseKey.

Si quieren córranlo, presionen el botón n.2 o el que dice “Nuevo Valor”, y después de presionarlo se meten en el registro presionan F5 para refrescar, buscan la carpeta en HKEY_CURRENT_USER, y le dan un click en la carpeta “PruebaAPI”, y verán en la columna derecha los dos valores que insertamos.

RegQueryValueEx
Api: Private Declare Function RegQueryValueEx Lib "advapi32.dll" Alias "RegQueryValueExA" (ByVal hKey As Long, ByVal lpValueName As String, ByVal lpReserved As Long, lpType As Long, lpData As Any, lpcbData As Long) As Long

Esta API se utiliza para recuperar un valor ya existente en el registro.

La explicación de los parámetro es igual a la del API RegSetValueEx con la única diferencia que “lpData” es la variable en donde almacenamos el “dato” y “lpcbData” es el tamaño de la variable. Se ve claro cuando expliquemos el ejemplo:

Private Sub Command3_Click()

Dim cadena As String

Dim binario As Long

cadena = String(255, Chr(0))

Dim res As Long

RegOpenKey HKEY_CURRENT_USER, "PruebaAPI", res

RegQueryValueEx res, "Valor String", 0, REG_SZ, ByVal cadena, Len(cadena)

RegQueryValueEx res, "Valor Binario", 0, REG_BINARY, binario, 4

MsgBox cadena

MsgBox binario

RegCloseKey res

End Sub

Aquí RegQueryValueEx como verán es exactamente igual a RegSetValueEx, con la diferencia que el quinto parámetro es la variable donde almacenamos el dato, en este caso usamos la variable “cadena”, y la variable “binario”, recuerdan que para pasar una cadena a una función API siempre hay que inicializarla asignándoles espacios en blanco en memoria, eso lo hacemos en la línea de código:

cadena = String(255, Chr(0))

Luego al final muestro el resultado y cierro el Registro.

RegDeleteValue
Api: Private Declare Function RegDeleteValue Lib "advapi32.dll" Alias "RegDeleteValueA" (ByVal hKey As Long, ByVal lpValueName As String) As Long
Esta API se utiliza para borrar un valor, y por consiguiente borra su dato. Los parámetros son sencillos y lo explico directo con el ejemplo que usamos en el Command4:

RegDeleteValue res, "Valor String"

El primer parámetro hKey es el Handler que nos devuelve RegOpenKey, y el segundo parámetro es el nombre del Valor. 

RegDeleteKey
Api: Private Declare Function RegDeleteKey Lib "advapi32.dll" Alias "RegDeleteKeyA" (ByVal hKey As Long, ByVal lpSubKey As String) As Long

Esta API se utiliza para borrar una clave y por consiguiente borra sus valores. Al igual que la anterior la explico con el ejemplo:

RegDeleteKey HKEY_CURRENT_USER, "PruebaAPI"

Como verán en el ejemplo aquí ni abrimos, ni cerramos el API por que como no vamos a eliminar a un valor, no es necesario. Recuerden usamos el RegOpenKey cuando queremos manipular el interior de una carpeta o clave.

Sus dos parámetros son el “hKey” que en este caso es valor de la carpeta “principal” donde esta la clave o carpeta “PruebaAPI”, y como segundo parámetro colocamos el nombre de la clave o carpeta.

En estos ejemplos hice un uso muy sencillo de las API’S que manejan el registro de Windows, habrán notado que en su mayoría use las API como si fueran métodos y no funciones, ustedes para mayor seguridad la pueden usar como si fuesen funciones para verificar si la creación, consulta, asignación, etc, se hizo con éxito, para ello sigan el ejemplo del Command1, también coloquen la carpeta adentro de HKEY_CURRENT_USER. Es recomendable, colocar la carpeta que queremos crear dentro de una carpeta que se encuentre en HKEY_CURRENT_USER, por ejemplo, los valores que usa nuestro programa, es recomendable colocar la clave o carpeta dentro de HKEY_CURRENT_USER/Software. Para ser eso, lo que cambia es  la clave, en ves de usar:

RegCreateKey(HKEY_CURRENT_USER, "PruebaAPI", res)

Usamos:

RegCreateKey(HKEY_CURRENT_USER, "SOFTWARE\PruebaAPI", res)

Así mantenemos un orden adentro del registro. A continuación, les muestro un ejemplo de cómo hacer que un programa arranque automáticamente cuando entre a Windows.

Para ello se necesita un formulario con dos Botones, esta aplicación no la voy a explicar, ya que es un ejemplo de lo que se puede hacer conociendo el Registro de Windows. En este caso el programa hace que se ejecute la calculadora cuando se reinicie Windows.

Const REG_SZ = 1

Const REG_BINARY = 3

Const HKEY_LOCAL_MACHINE = &H80000002

Private Declare Function RegCloseKey Lib "advapi32.dll" (ByVal hKey As Long) As Long

Private Declare Function RegCreateKey Lib "advapi32.dll" Alias "RegCreateKeyA" (ByVal hKey As Long, ByVal lpSubKey As String, phkResult As Long) As Long

Private Declare Function RegDeleteValue Lib "advapi32.dll" Alias "RegDeleteValueA" (ByVal hKey As Long, ByVal lpValueName As String) As Long

Private Declare Function RegSetValueEx Lib "advapi32.dll" Alias "RegSetValueExA" (ByVal hKey As Long, ByVal lpValueName As String, ByVal Reserved As Long, ByVal dwType As Long, lpData As Any, ByVal cbData As Long) As Long

Private Sub Command1_Click()

    SaveString HKEY_LOCAL_MACHINE, "Software\Microsoft\Windows\CurrentVersion\Run", "Cal", "c:\winnt\system32\calc.exe"

End Sub

Private Sub Command2_Click()

    DelSetting HKEY_LOCAL_MACHINE, "Software\Microsoft\Windows\CurrentVersion\Run", "Cal"

End Sub

Private Sub Form_Load()

Command1.Caption = "PONER CALCULADORA"

Command2.Caption = "QUITAR CALCULADORA"

End Sub

Sub SaveString(hKey As Long, strPath As String, strValue As String, strData As String)

    Dim Ret

    RegCreateKey hKey, strPath, Ret

    RegSetValueEx Ret, strValue, 0, REG_SZ, ByVal strData, Len(strData)

    RegCloseKey Ret

End Sub

Sub DelSetting(hKey As Long, strPath As String, strValue As String)

    Dim Ret

    RegCreateKey hKey, strPath, Ret

    RegDeleteValue Ret, strValue

    RegCloseKey Ret

End Sub

Con lo explicado arriba estoy seguro que entenderán el código.

OTRA MANERA DE VER LOS ARCHIVOS

Ya hemos aprendido la opción de almacenar información en el registro de Windows, ahora veremos la segunda forma la cual consta de la utilización de los archivos .INI estos archivos presentan una estructura similar a la que expongo a continuación:

[NOMBRE]

Valor1=Dato1

Valor2=Dato2

Valor3=Dato3

Aquí NOMBRE es la identificación del cuerpo que contendrá “VALORES”, y a su vez estos “valores” contendrán sus “datos”, el manejo de estas API’S son muchos mas sencilla que la del registro. A continuación pongo un ejemplo hecho en VB y luego su respectiva explicación: (Para este código se necesitan 2 botones)

Private Declare Function GetPrivateProfileString Lib "kernel32" Alias "GetPrivateProfileStringA" (ByVal lpApplicationName As String, ByVal lpKeyName As Any, ByVal lpDefault As String, ByVal lpReturnedString As String, ByVal nSize As Long, ByVal lpFileName As String) As Long

Private Declare Function WritePrivateProfileString Lib "kernel32" Alias "WritePrivateProfileStringA" (ByVal lpApplicationName As String, ByVal lpKeyName As Any, ByVal lpString As Any, ByVal lpFileName As String) As Long

Dim strCadena As String

Dim resp As Long

Private Sub Command1_Click()

    WritePrivateProfileString "CUERPO", "VALOR1", "Hola Mundo", "c:\ejemplo.ini"

    WritePrivateProfileString "CUERPO", "VALOR2", "Hola Mundo 2", "c:\ejemplo.ini"

    WritePrivateProfileString "CUERPO 2", "VALOR1", "Hola Mundo 3", "c:\ejemplo.ini"

End Sub

Private Sub Command2_Click()

    strCadena = String(255, 0)

    resp = GetPrivateProfileString("CUERPO", "VALOR1", "Default", strCadena, 255, "C:\ejemplo.ini")

    If resp <> 0 Then strCadena = Left$(strCadena, resp)

    MsgBox strCadena

    strCadena = String(255, 0)

    resp = GetPrivateProfileString("CUERPO", "VALOR2", "Default", strCadena, 255, "C:\ejemplo.ini")

    If resp <> 0 Then strCadena = Left$(strCadena, resp)

    MsgBox strCadena

    strCadena = String(255, 0)

    resp = GetPrivateProfileString("CUERPO 2", "VALOR1", "Default", strCadena, 255, "C:\ejemplo.ini")

    If resp <> 0 Then strCadena = Left$(strCadena, resp)

    MsgBox strCadena

End Sub

Private Sub Form_Load()

Command1.Caption = "Guardar Datos"

Command2.Caption = "Recuperar Datos"

End Sub

Mas fácil y mas sencillo no podía ser. Ahora vamos a explicar cada función API utilizada:

WritePrivateProfileString

Api: Declare Function WritePrivateProfileString Lib "kernel32" Alias "WritePrivateProfileStringA" (ByVal lpApplicationName As String, ByVal lpKeyName As Any, ByVal lpString As Any, ByVal lpFileName As String) As Long

Esta función API es la que se utiliza para escribir en el ARCHIVO. Si el archivo no existe la función API se encarga de crearlo y si existe lo sobrescribe.

NOTA: La mayoría de la gente cree que esta función es para escribir en extensiones .INI, y están en lo cierto pero ustedes pueden usar cualquier extensión y las dos funciones funcionarán perfectamente siempre cuando respeten la estructura que mencione arriba. Dado a que esa estructura se presenta en su mayoría en los archivos .INI, no significa que esta función sea solamente para ellos. Si quieren hagan la prueba en el ejemplo de arriba pongan en nombre del archivo “ejemplo.ttt” y verán que funcionará exactamente igual como los .INI, “repito” lo único que tienes que respetar es la estructura que coloque arriba.

Entre sus parámetros tenemos:

lpApplicationName: No es mas, que el nombre del cuerpo de los valores, se recuerdan de la estructura que puse arriba  “lpApplicationName” es “[NOMBRE]”

lpKeyName = Es el nombre del Valor

lpString = Es el dato que contendrá el “Valor” o lpKeyName.

lpFileName =  Es la ruta y el nombre del archivo donde se guardará la información.

En el ejemplo de COMMAND1_CLICK tenemos

WritePrivateProfileString "CUERPO", "VALOR1", "Hola Mundo", "c:\ejemplo.ini"

WritePrivateProfileString "CUERPO", "VALOR2", "Hola Mundo 2", "c:\ejemplo.ini"

WritePrivateProfileString "CUERPO 2", "VALOR1", "Hola Mundo 3", "c:\ejemplo.ini"

Aquí creo dos Cuerpos, el primer cuerpo cuyo nombre es “Cuerpo” tiene dos valores con sus respectivos datos. Y el segundo cuerpo “Cuerpo 2” solamente le inserte un solo valor. Yo creo que esta API no tiene complicación si quieren después de crear el archivo, abran el archivo por el Bloc de Notas y vean la estructura que se creo.

GetPrivateProfileString 

Api: Declare Function GetPrivateProfileString Lib "kernel32" Alias "GetPrivateProfileStringA" (ByVal lpApplicationName As String, ByVal lpKeyName As Any, ByVal lpDefault As String, ByVal lpReturnedString As String, ByVal nSize As Long, ByVal lpFileName As String) As Long
Esta función API como sus primeras tres letras lo dicen GET “capturar o recuperar”, recuperamos el Datos que presenta un valor, que se encuentre en un archivo .INI o .XXX extensión.

Entre sus parámetros tenemos:

lpApplicationName: la misma explicación del API “WritePrivateProfileString”

 lpKeyName = Es el nombre del Valor

lpDefault = Es un valor “String”, el cual se usa de la siguiente manera, si “lpKeyName” no existe en el archivo  o no esta adentro de “lpApplicationName”, la cadena string que regresa es la que se coloque aquí. Por ejemplo si en este caso:

    strCadena = String(255, 0)

    resp = GetPrivateProfileString("CUERPO 2", "VALOR1", "Default", strCadena, 255, "C:\ejemplo.ini")

    If resp <> 0 Then strCadena = Left$(strCadena, resp)

    MsgBox strCadena

Cambiamos “VALOR1” por “VALORX”, como “VALORX” no esta en ese cuerpo, la variable “strCadena” retorna el valor por defecto en este caso “Default”.

lpReturnedString =  Es la variable donde se almacenará la cadena String que la función retorne.

nSize = Es el tamaño de caracteres que queremos retornar

lpFileName =  Es la ruta y el nombre del archivo donde esta la información.

En el ejemplo con COMMAND2_CLICK():

    strCadena = String(255, 0)

    resp = GetPrivateProfileString("CUERPO", "VALOR1", "Default", strCadena, 255, "C:\ejemplo.ini")

    If resp <> 0 Then strCadena = Left$(strCadena, resp)

    MsgBox strCadena

Lo que hacemos es, primero como les he dicho siempre, toda API que retorne un STRING, la variable donde se va a almacenar el valor, debe de ser inicializada    “strCadena = String(255, 0)”, luego ejecutamos “GetPrivateProfileString” con los parámetros que ya explique. Esta API retorna la longitud de la cadena que fue almacenada en “strCadena”, esta longitud se encuentra en la variable “resp”, por esa razón utilizamos “Left$(strCadena, resp)”, como la variable “strCadena” fue inicializada con 255 espacios en blancos, con la función de VB “Left”, lo que hacemos es quitar los espacios en blancos que son innecesarios. Y luego por último mostramos la cadena.

Ya con esto, les enseñe los dos formas de guardar información de manera ordenada. En ustedes quedará la manera de cómo utilizarlo. Personalmente, yo utilizo los dos, de la siguiente manera. Cuando mi programa necesita valores iniciales para su arranque, normalmente los coloco en el registro de WINDOWS, ahora si necesito almacenar muchos valores es ahí donde utilizo los Archivos .INI. Pero en ustedes quedará el como usar para su programa cada uno de los métodos.

MENUS

El menú es un control que VB simplificó de una buena manera, VB usa un editor de menú para crear los menús a los formularios. Pero algunas veces nace la necesidad de crear POPMENUS en controles que no tienen esta propiedad, o también la colocación de imágenes al lado de los textos del menú. Entonces es aquí donde entra en funcionamiento las API’S de WINDOW. En el ejemplo que voy a mostrar aumento un poco la dificultad, en la manera de cómo se usan las API’S, inclusive vamos a capturar eventos que están fuera de los eventos tradicionales que presentan los controles.

Para este proyecto se necesitan varias cosas:

1. Dos Formulario

2. Un Modulo

3. En el Form1 pongan 3 botones, un textbox, un listbox, un timer, un imagelist y un picture (En el imagelist deberán de colocar 3 imágenes de 16x16, y en el picture colocan una imagen de la misma dimensión 16x16. Coloquen la propiedad “Autosize” en True).

4. En el Form1 creen un menú con dos submenús llamen el primero “Archivo” y el segundo “Edicion”, dentro de “Archivo” pongan 3 items, y dentro de “Edicion” pongan dos items.

5. En el Form2 creen un menú con dos submenús llamen el primero “TextBox” y el segundo “Listbox”, dentro de “TextBox” pongan 3 Item, y dentro de “ListBox” pongan dos Item.

Hecho todo esto al pie de la letra, copien el código que les presento a continuación:

En el modulo:

Option Explicit

Declare Function SetWindowLong Lib "user32" Alias "SetWindowLongA" (ByVal hwnd As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As Long

Declare Function CallWindowProc Lib "user32" Alias "CallWindowProcA" (ByVal lpPrevWndFunc As Long, ByVal hwnd As Long, ByVal Msg As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

Public Const WM_NCRBUTTONDOWN = &HA4

Public Const WM_NCLBUTTONDOWN = &HA1

Public Const WM_NCXBUTTONDOWN = &HAB

Public Const WM_XBUTTONDOWN = &H20B

Public Const BN_CLICKED = 0

Public Const WM_NCMOUSEMOVE = &HA0

Public Const GWL_WNDPROC = -4

Public Const WM_COMMAND = &H111

Public nuevoproc As Long

Public viejoproc As Long

Public mensageForm As String

Sub Cambiar_Proc(hwnd2 As Long)

nuevoproc = ObtenerLong(AddressOf MenuProc)

viejoproc = SetWindowLong(hwnd2, GWL_WNDPROC, nuevoproc)

End Sub

Function ObtenerLong(ByVal direcfunt As Long) As Long

    ObtenerLong = direcfunt

End Function

Private Function MenuProc(ByVal hwnd As Long, ByVal message As Long, ByVal wParam As Long, ByVal lParam As Long) As Long     

    If message = WM_NCLBUTTONDOWN Then

        If wParam = 8 Then

            MsgBox "POR QUE ME MINIMIZAS"

        End If

        If wParam = 9 Then

            MsgBox "MAXIMIZAR YES!!!!!!!"

        End If

        If wParam = 20 Then

            MsgBox "POR QUE ME QUIERES CERRAR?!!!!!"

        End If

    End If

    If message = WM_NCMOUSEMOVE Then

        If wParam = 5 Then mensageForm = "ENTRASTES AL MENU!!!!"

        If wParam = 2 Then mensageForm = "OJO: ENTRASTES A ZONA PROHIBIDA!!!!"

    End If

    If message = WM_COMMAND Then

    If lParam = 0 Then

        Select Case wParam

            Case 0

                MsgBox "Le doy a la opción 1"

            Case 1

                MsgBox "Le doy a la opción 2"

        End Select

    End If

    End If

    MenuProc = CallWindowProc(viejoproc, hwnd, message, wParam, lParam)

End Function

En el Form1:

Const TPM_LEFTALIGN = &H0&

Const MF_STRING = &H0&

Const MF_BYPOSITION = &H400&

Const MF_REMOVE = &H1000&

Const MF_DISABLED = &H2&

Const MF_BITMAP = &H4&

Const TPM_RIGHTALIGN = &H8&

Private Type POINTAPI

    X As Long

    Y As Long

End Type

Private Declare Function CreatePopupMenu Lib "user32" () As Long

Private Declare Function TrackPopupMenu Lib "user32" (ByVal hMenu As Long, ByVal wFlags As Long, ByVal X As Long, ByVal Y As Long, ByVal nReserved As Long, ByVal hwnd As Long, ByVal lprc As Any) As Long

Private Declare Function GetSystemMenu Lib "user32" (ByVal hwnd As Long, ByVal bRevert As Long) As Long

Private Declare Function AppendMenu Lib "user32" Alias "AppendMenuA" (ByVal hMenu As Long, ByVal wFlags As Long, ByVal wIDNewItem As Long, ByVal lpNewItem As Any) As Long

Private Declare Function DestroyMenu Lib "user32" (ByVal hMenu As Long) As Long

Private Declare Function GetCursorPos Lib "user32" (lpPoint As POINTAPI) As Long

Private Declare Function GetMenu Lib "user32" (ByVal hwnd As Long) As Long

Private Declare Function GetSubMenu Lib "user32" (ByVal hMenu As Long, ByVal nPos As Long) As Long

Private Declare Function GetMenuItemID Lib "user32" (ByVal hMenu As Long, ByVal nPos As Long) As Long

Private Declare Function GetMenuItemCount Lib "user32" (ByVal hMenu As Long) As Long

Private Declare Function RemoveMenu Lib "user32" (ByVal hMenu As Long, ByVal nPosition As Long, ByVal wFlags As Long) As Long

Private Declare Function DrawMenuBar Lib "user32" (ByVal hwnd As Long) As Long

Private Declare Function ModifyMenu Lib "user32" Alias "ModifyMenuA" (ByVal hMenu As Long, ByVal nPosition As Long, ByVal wFlags As Long, ByVal wIDNewItem As Long, ByVal lpString As Any) As Long

Private Declare Function SetMenuItemBitmaps Lib "user32" (ByVal hMenu As Long, ByVal nPosition As Long, ByVal wFlags As Long, ByVal hBitmapUnchecked As Long, ByVal hBitmapChecked As Long) As Long

Dim hMenu As Long

Dim hMenuSystem As Long

Dim bool As Boolean

Private Sub Command1_Click()

If bool Then

    Dim Numero As Long

    hMenuSystem = GetSystemMenu(Me.hwnd, False)

    Numero = GetMenuItemCount(hMenuSystem)

    RemoveMenu hMenuSystem, Numero - 1, MF_BYPOSITION Or MF_DISABLED

    DrawMenuBar Me.hwnd

    bool = False

End If

End Sub

Private Sub Command2_Click()

    ModifyMenu GetSubMenu(GetMenu(Me.hwnd), 0), GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), 0), MF_BITMAP, GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), 0), CLng(ImageList1.ListImages(1).Picture)

    ModifyMenu GetSubMenu(GetMenu(Me.hwnd), 0), GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), 1), MF_BITMAP, GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), 1), CLng(ImageList1.ListImages(2).Picture)

    ModifyMenu GetSubMenu(GetMenu(Me.hwnd), 0), GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), 2), MF_BITMAP, GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), 2), CLng(ImageList1.ListImages(3).Picture)

End Sub

Private Sub Command3_Click()

    SetMenuItemBitmaps GetSubMenu(GetMenu(Me.hwnd), 1), 0, MF_BYPOSITION, Picture1.Picture, Picture1.Picture

End Sub

Private Sub Form_Load()

    Command1.Caption = "Anular el botón 'X' de Salir"

    Command2.Caption = "Imagenes en Menu (1)"

    Command3.Caption = "Imagenes en Menu (2)"

    Cambiar_Proc (Me.hwnd)

    hMenu = CreatePopupMenu()

    AppendMenu hMenu, MF_STRING, ByVal 0&, "Opcion 1"

    AppendMenu hMenu, MF_SEPARATOR, ByVal 1&, ByVal 0&

    AppendMenu hMenu, MF_STRING, ByVal 1&, "Opcion 3"

    bool = True

End Sub

Private Sub Form_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)

If Button = 2 Then

    Dim Puntero As POINTAPI

    GetCursorPos Puntero

    TrackPopupMenu hMenu, TPM_LEFTALIGN, Puntero.X, Puntero.Y, 0, Me.hwnd, ByVal 0&

End If

End Sub

Private Sub Form_MouseMove(Button As Integer, Shift As Integer, X As Single, Y As Single)

    mensageForm = "FORMULARIO"

End Sub

Private Sub Form_Unload(Cancel As Integer)

    DestroyMenu hMenu

End Sub

Private Sub mnuSalir_Click()

        Unload Me

End Sub

Private Sub List1_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)

If Button = 2 Then

    Dim Puntero As POINTAPI

    GetCursorPos Puntero

    TrackPopupMenu GetSubMenu(GetMenu(Form2.hwnd), 1), TPM_LEFTALIGN, Puntero.X, Puntero.Y, 0, Form2.hwnd, ByVal 0&

End If

End Sub

Private Sub Text1_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)

If Button = 2 Then

    Dim Puntero As POINTAPI

    GetCursorPos Puntero

    TrackPopupMenu GetSubMenu(GetMenu(Form2.hwnd), 0), TPM_LEFTALIGN, Puntero.X, Puntero.Y, 0, Form2.hwnd, ByVal 0&

End If

End Sub

Private Sub Timer1_Timer()

    Cls

    Print mensageForm

End Sub

Para empezar, creo que será conveniente explicar ¿Que hace el modulo?. 

Hay que entender ciertas cosas primeros. Cuando ustedes realizan una acción, Windows genera un Mensaje, esto lo vimos en la parte donde se explicaba el SendMessage, ese mensaje se pasea por todas las aplicaciones, hasta llegar a su objetivo final. Este “objetivo”, (que puede ser una ventana, control, clase, etc), tiene que tener una función de procedimiento asociado que es la que se encarga de procesar el mensaje que envió Windows. Es decir, cuando ustedes corren el formulario de VB y le dan click sobre él. Se genera un mensaje, ese mensaje esta compuesto por:

El “hwnd”, por un “Message”, luego “wParam” y por último “lParam”, pero el formulario internamente tiene una función que le indica a Windows lo siguiente. Por ejemplo. Windows envía el mensaje. El mensaje llega al formulario y busca cual es la función de procedimiento asociada a los mensajes (Nota: no es una función por cada mensaje, no! Una sola función los controla a todos). Esto es transparente en VB. Los programadores no deciden ni intervienen en nada de esto. Solamente cuando programas bajo Windows es que lo ves.

Ahora en el modulo tengo tres funciones API las cuales voy a explicar:

SetWindowLong

Api: Declare Function SetWindowLong Lib "user32" Alias "SetWindowLongA" (ByVal hwnd As Long, ByVal nIndex As Long, ByVal dwNewLong As Long) As Long

Esta función cambia o asigna los atributos de una ventana.

Entre sus parámetro tenemos a:

hwnd =  Bueno creo que ya esto lo tenemos claro, el hwnd del objeto o ventana a la cual queremos modificarle su atributo.

nIndex = Es la constante que determina que atributo se va a asignar o cambiar:

Public Const GWL_EXSTYLE = (-20)

Public Const GWL_HINSTANCE = (-6)

Public Const GWL_STYLE = (-16)

Public Const GWL_WNDPROC = (-4)

En  nuestro caso usamos GWL_WNDPROC.

dwNewLong = Es la dirección del nuevo atributo.

La función retorna un valor long que identifica  la función anterior que había ahí.

Ahora como lo aplicamos a nuestro programa, aquí tenemos dos funciones:

Sub Cambiar_Proc(hwnd2 As Long)

nuevoproc = ObtenerLong(AddressOf MenuProc)

viejoproc = SetWindowLong(hwnd2, GWL_WNDPROC, nuevoproc)

End Sub

Function ObtenerLong(ByVal direcfunt As Long) As Long

    ObtenerLong = direcfunt

End Function

En Cambiar_Proc, le asignamos con SetWindowLong un nuevo procedimiento a la ventana o hwnd enviado. Pero antes de realizar esta llamada hacemos una:

nuevoproc = ObtenerLong(AddressOf MenuProc)

Como le dije, donde se procesan los mensaje es en una función, pero para asignársela al hwnd tenemos que pasarle la dirección de memoria donde esta la nueva función. Eso lo hacemos es con “AddressOf” esta función de VB me retorna la dirección de memoria de la función MenuProc la cual esta en la parte inferior del modulo. Uso otra función ObtenerLong lo cual por lo que deduzco se utiliza para retornar la dirección pero en una variable “long” o de 32Bit. Ya que el AddressOf no se puede utilizar de manera directa, es decir:

nuevoproc = AddressOf MenuProc

Luego que tenemos la dirección ya lo podemos asignar al hwnd. Esto lo conseguimos con:

viejoproc = SetWindowLong(hwnd2, GWL_WNDPROC, nuevoproc)

Con GWL_WNDPROC le decimos que cambie o asigne otra función capturadora de mensajes al hwnd, siendo la nueva función “nuevoproc”. Y listo! cuando el hwnd genere o llegue un mensaje esta función “MenuProc” lo podrá procesar.

CallWindowProc

Api: Declare Function CallWindowProc Lib "user32" Alias "CallWindowProcA" (ByVal lpPrevWndFunc As Long, ByVal hwnd As Long, ByVal Msg As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

función que permite pasarle información de un mensaje, a una función de procedimiento que este en un hwnd.

parámetros:

lpPrevWndFunc =  dirección de la función que queremos llamar.

hwnd = Handle de la ventana

El resto de los parámetro es la misma explicación que usamos en la función API SendMessage.

En la función MenuProc, tenemos esta línea:

MenuProc = CallWindowProc(viejoproc, hwnd, message, wParam, lParam) 

¿Por que la línea?. Lo que ocurre es que cuando invocamos a SetWindowLong en la función de arriba le estamos indicando a Windows que los mensajes de la ventana ya no los procesara la función de procedimientos Estándar que VB coloca en el formulario (Por que en este caso estamos es cambiando la función de procedimiento del Form1 esto lo veremos cuando entremos al código del formulario). Ya que la suplantamos por la nuestra  MenuProc. Esto produce que las instrucciones de pintado y todo el código del formulario que esta en los eventos de esa función de procedimiento no se ejecuten. Por esa razón  CallWindowProc lo que hace es llamar a esa función predeterminada del formulario pasándole los parámetros de un Mensaje de Windows (El mensaje de Windows se compone básicamente del hwnd, message,wparam y lparam) Así logramos que el formulario escuche los mensajes y eventos con la función standard, y la función MenuProc.

Se que todo esto suena complicado, hasta para mi si hace complicado de explicar, pero en resumen en el modulo hacemos esto: 

1.Toda ventana, control, o Handle posee una función de procedimientos que es la que se encarga de procesar los mensajes.

2.Podemos cambiar esa función de procedimientos con SetWindowLong

3.Pero cuando cambiamos la función de procedimiento es recomendable regresar o llamar a la función de procedimientos original, esto lo logramos con CallWindowProc.

De todas maneras no se preocupen si no lo entendieron, cuando veamos el código de Form1 si lo van a entender.

Antes de continuar corran el programa, vean el menú de manera original y luego presionen lo botones para incluir las imágenes, vuelvan a ver el menú y vean los cambios. Luego presionen el botón de “Anular ‘X’ de Salir”, y notarán que el botón ‘X’ de la ventana se deshabilita. Luego pasen el mouse por el menú y verán un mensaje que dice “Entrastes al menú”, a su vez suban el puntero hasta el “Caption” de la ventana y verán que el mensaje cambia a “Entrastes a la Zona Prohibida”. También presionen el botón secundario del mouse en el formulario para que vean un POPMENU, y por último presionen el botón secundario tanto el TextBox como en el ListBox.  Esto es bueno que lo hagan primero para que entiendan lo que voy a explicar a continuación.

CreatePopupMenu

Api: Private Declare Function CreatePopupMenu Lib "user32" () As Long

Esta API se utiliza para crear un menú en blanco o “Empty”. Como notarán no presenta ningún parámetro. En el formulario se encuentra con esta instrucción:

hMenu = CreatePopupMenu()

AppendMenu

Api: Private Declare Function AppendMenu Lib "user32" Alias "AppendMenuA" (ByVal hMenu As Long, ByVal wFlags As Long, ByVal wIDNewItem As Long, ByVal lpNewItem As Any) As Long

Esta función es la que se encarga de añadir nuevos ítem, al menú. Sus parámetros son los siguientes:

hMenu = Es el valor que retorna CreatePopupMenu()

wFlags = Son indicadores que especifican el estado o que tipo de ítem es. Por ejemplo si el menú es un Check hay un Flags para indicarlo.


Const MF_CHECKED = &H8& ‘ El item es un Check

Const MF_GRAYED = &H1& 

Const MF_SEPARATOR = &H800& ‘ Es la línea separadora 

Const MF_STRING = &H0& ‘ Es una cadena de Texto

wIDNewItem = Es el número de ITEM o el número que lo identifica es muy importante no repetir este número ya que como veremos mas adelante esto es lo que los diferencian unos de otros.

lpNewItem = Colocan la cadena de texto que quieren mostrar.

En el formulario usamos el API de la siguiente manera:

AppendMenu hMenu, MF_STRING, ByVal 0&, "Opcion 1"

AppendMenu hMenu, MF_SEPARATOR, ByVal 1&, ByVal 0&

AppendMenu hMenu, MF_STRING, ByVal 1&, "Opcion 3"

Con esto creamos 3 ítem de un menú, el primero es String cuyo texto es “opción 1”, el segundo es una línea divisoria, y el tercero es un STRING. Creo que aquí no hay gran explicación. Pero ustedes dirán ahora ¿Cómo mostramos el menú?.

TrackPopupMenu

Api: Private Declare Function TrackPopupMenu Lib "user32" (ByVal hMenu As Long, ByVal wFlags As Long, ByVal X As Long, ByVal Y As Long, ByVal nReserved As Long, ByVal hwnd As Long, ByVal lprc As Any) As Long
Esta es el API que se utiliza para mostrar un menú. Sus parámetros son:

hMenu =  el valor que retorna CreatePopupMenu()

wFlags = Aquí lo Flags determinan como se va a mostrar el menú. Sus constantes son:

Public Const TPM_BOTTOMALIGN = &H0020&

Public Const TPM_CENTERALIGN = &H4&

Public Const TPM_HORIZONTAL = &H0000&

Public Const TPM_LEFTALIGN = &H0&

Public Const TPM_RIGHTALIGN = &H8&

Public Const TPM_TOPALIGN = &H0000&

Public Const TPM_VCENTERALIGN = &H0010&

Public Const TPM_VERTICAL = &H0040&

X = Coordenada del eje de las X donde mostrara el menú.

Y = Coordenada del eje de las Y donde mostrara el menú.

nReserved = Siempre se coloca 0

hwnd = Este es el hwnd de la ventana a la cual se asocia el menú

lprc = Se coloca 0.

En el formulario lo usamos así, en el evento MouseDown:

Private Sub Form_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)

If Button = 2 Then

    Dim Puntero As POINTAPI

    GetCursorPos Puntero

    TrackPopupMenu hMenu, TPM_LEFTALIGN, Puntero.X, Puntero.Y, 0, Me.hwnd, ByVal 0&

End If

End Sub

Lo que hacemos es “SI el BOTON es el SECUNDARIO (Button = 2) entonces captura el eje del puntero y muéstrame el menú”, ya voy a explicar que hace “GetCursorPos” pero creo que ya deberían de tener una idea. Y los parámetros de “TrackPopupMenu” creo que también están claros.

GetCursosPos

API: Private Declare Function GetCursorPos Lib "user32" (lpPoint As POINTAPI) As Long

API utilizada para capturar el eje de las X y Y en donde se encuentre el puntero en ese momento. Recibe como único parámetro una estructura de tipo POINTAPI. Que es la siguiente:

Private Type POINTAPI

    X As Long

    Y As Long

End Type

Esta estructura se llena cuando se le pasa a la función:

    GetCursorPos Puntero

Con esto obtengo el eje X y Y, de donde se encuentra el puntero en ese momento y es ahí en donde despliego, en el caso de Form1, mi menú.

DestroyMenu

API: Private Declare Function DestroyMenu Lib "user32" (ByVal hMenu As Long) As Long

Esta API destruye el menú de la memoria. Esto es muy importante, si no destruyen el menú VB le puede generar un error que los saque del programa. Es vital que en el UNLOAD del formulario principal destruyan los menú creados. En nuestro caso es uno solo y el código es el siguiente:

Private Sub Form_Unload(Cancel As Integer)

    DestroyMenu hMenu

End Sub

Y creo que el parámetro esta mas que claro.

Mensajes al menú

Ahora ustedes se preguntarán ¿Cómo capturo el mensaje del menú? Ya que cuando hacemos un menú en VB, él se encarga de dibujarlo y cuando uno le da click a nivel de diseño, te abre la pantalla de código para que tu coloques el algoritmo a seguir. Pero en este caso estas creando un MENU bajo código así que ¿Cómo se hace?. Lo explico a continuación:

Se recuerdan de la explicación del Modulo, bueno si nos les quedo claro espero que después de aquí logren entenderlo:

Primero veamos esta línea que esta en el evento LOAD de Form1.

    Cambiar_Proc (Me.hwnd)

Aquí llamo a la función de Cambiar_Proc que esta en el modulo, y es el que se encarga de cambiar la función de mensajes del Formulario Form1. Es decir en este momento le dijo a Windows “Olvídate de los la función de procedimiento viejo y has caso a la función MenuProc”. En este momento cuando Windows envía un mensaje, pasa primero por MenuProc y luego gracias a la función API CallWindowProc pasa por la función estándar del Formulario. Si no ponemos CallWindowProc su aplicación no servirá, pruébenlo y verán.

Ahora veamos que hay en la función:

Private Function MenuProc(ByVal hwnd As Long, ByVal message As Long, ByVal wParam As Long, ByVal lParam As Long) As Long

    If message = WM_NCLBUTTONDOWN Then

        If wParam = 8 Then

            MsgBox "POR QUE ME MINIMIZAS"

        End If        

        If wParam = 9 Then

            MsgBox "MAXIMIZAR YES!!!!!!!"

        End If        

        If wParam = 20 Then

            MsgBox "POR QUE ME QUIERES CERRAR?!!!!!"

        End If

    End If

    If message = WM_NCMOUSEMOVE Then

        If wParam = 5 Then mensageForm = "ENTRASTES AL MENU!!!!"

        If wParam = 2 Then mensageForm = "OJO: ENTRASTES A ZONA PROHIBIDA!!!!"

    End If

    If message = WM_COMMAND Then    

    If lParam = 0 Then

        Select Case wParam

            Case 0

                MsgBox "Le doy a la opción 1"

            Case 1

                MsgBox "Le doy a la opción 2"

        End Select

    End If

    End If

    MenuProc = CallWindowProc(viejoproc, hwnd, message, wParam, lParam)

End Function

En este momento olvídense de los mensajes iniciales, vamos a la línea WM_COMMAND, cuando ustedes presionan un botón, menú, etc. Windows genera un mensaje COMMAND y lo envía a la aplicación pertinente, como el mensaje primero llega aquí y luego a la función predeterminada del formulario yo lo capturo con el If.

    If message = WM_COMMAND Then    

Ahora como los botones, el menú, y otros controles generan el mensaje, tengo que decirle que solamente ejecute los de menú que yo cree. Esto lo hago con la línea:

    If lParam = 0 Then

Y luego con wParam veo a que ítem del menú le estoy dando click, se recuerdan del parámetro del API AppendMenu “wIDNewItem”, pues wParam contiene ese valor, para indicar a que ítem se presiono. Y tanto para el ítem 0 como para el 1 hay un mensaje distinto. Vieron que es sencillo!!!. Ahora, ya que estamos aquí adentro vamos a explicar los otros mensajes.

    If message = WM_NCMOUSEMOVE Then

        If wParam = 5 Then mensageForm = "ENTRASTES AL MENU!!!!"

        If wParam = 2 Then mensageForm = "OJO: ENTRASTES A ZONA PROHIBIDA!!!!"

    End If

Este mensaje no se puede capturar con los eventos tradicionales de VB, tienes que irte por estos métodos para capturarlos. Cuando el puntero esta en una zona del formulario que no es el área cliente. Llamo área cliente a nada mas la parte gris del formulario. Todo lo que es menú y el Caption es área de “no cliente”. Es por esa razón que el mensaje que se usa es WM_NCMOUSEMOVE, si fuera en el área cliente el mensaje seria WM_MOUSEMOVE, si ven un mensaje WM_NC significa que es un mensaje que se genero en un área “no cliente”. En VB los eventos son orientados al área cliente. Por esa razón para lograr esto, se tienen que ir por estos caminos, engorrosos y turbios de la programación. wParam se encarga de decirme por donde esta pasando si es un menú, wParam se hace igual a 5, y si es la barra superior o el Caption de la ventana wParam es igual a 2.

    If message = WM_NCLBUTTONDOWN Then

        If wParam = 8 Then

            MsgBox "POR QUE ME MINIMIZAS"

        End If        

        If wParam = 9 Then

            MsgBox "MAXIMIZAR YES!!!!!!!"

        End If        

        If wParam = 20 Then

            MsgBox "POR QUE ME QUIERES CERRAR?!!!!!"

        End If

    End If

Cabe decir, que con VB no podemos capturar el evento click de los botones de la ventana, me refiero a los 3 de la esquina superior derecha (Minimizar, Maximizar, y Cerrar). En VB lo que se hace es capturar el estado de la pantalla. Pero no capturas el click. Con el mensaje WM_NCLBUTTONDOWN vean que tiene NC, esto es por que los botones están en un área de “no cliente”, luego usamos wParam que nos dice, que cuando aprietas el de minimizar wParam es igual a 8, para maximizar wParam = 9 y para cerrar wParam = 20, esa es la razón del mensaje cuando ustedes aprietan algunos de los botones.

Recuerden algo, Windows tiene un centenar de mensajes y VB no captura todos, obviamente VB captura solo los mas usados e importantes pero uno nunca sabe cuando puede utilizar uno de estos.

IMÁGENES EN EL MENU

Una de las cosas, que más buscan en la creación de menús es el colocar imágenes al lado del texto. Esto lo logramos de dos maneras, podemos usar menús completamente gráficos, o usar el tradicional menú de texto con una imagen a lado. A continuación la explicación de estos dos métodos:

GetMenu

Api: Private Declare Function GetMenu Lib "user32" (ByVal hwnd As Long) As Long

Esta API retorna el handle del menú que usamos en una ventana. Como único parámetro tenemos el hwnd de la ventana que contiene el menú.

GetSubMenu

Api: Private Declare Function GetSubMenu Lib "user32" (ByVal hMenu As Long, ByVal nPos As Long) As Long
Esta API retorna el handle de un SUB-MENU del menú principal. Como parámetros tenemos:

hMenu = Es el handle del menú principal

nPos = Es el número de índice del sub-menú. (El índice empieza desde 0).

Entiendan algo: “menú” es todo el menú que muestra VB en nuestro caso es Archivo y Edicion. “Sub-menú”, son los distintos enlaces del menú, es decir, Archivo y Edicion, e “ítem”, es lo que contiene los “Sub-menús”, en el caso de Archivo son:  Nuevo, Guardar, y Salir.

SetMenuItemBitmaps

Api: Private Declare Function SetMenuItemBitmaps Lib "user32" (ByVal hMenu As Long, ByVal nPosition As Long, ByVal wFlags As Long, ByVal hBitmapUnchecked As Long, ByVal hBitmapChecked As Long) As Long

Esta API se utiliza para colocar una imagen al lado izquierdo del texto del menú. Entre sus parámetro tenemos:

hMenu = El handle del menú.

nPosition = El número de ITEM en donde queremos posicionar la imagen.

wFlags = Coloquen MF_BYPOSITION.

hBitmapUnchecked = En caso de que el ITEM sea un Check. La imagen que coloquen aquí es la que mostrara cuando el ITEM no este activado.

hBitmapCheked = En caso de que el ITEM sea un Check. La imagen que coloquen aquí es la que mostrará cuando el ITEM este activado. Si su menú es de texto y no es un “Checked”, coloquen la misma imagen en los dos parámetros.

En el formulario lo usamos de manera muy simple:

Private Sub Command3_Click()

    SetMenuItemBitmaps GetSubMenu(GetMenu(Me.hwnd), 1), 0, MF_BYPOSITION, Picture1.Picture, Picture1.Picture

End Sub

Noten algo curioso:

GetSubMenu(GetMenu(Me.hwnd), 1)

Esto es igual a:

Varlong = GetMenu(Me.hwnd)

Varlong2 = GetSubMenu(Varlong,1)

Y luego:

SetMenuItemBitmaps Varlong2,…………..

Dos variables y tres líneas de códigos usamos, y la simplificamos en una sola “GetSubMenu(GetMenu(Me.hwnd), 1)”. Esto lo coloque para que vean las distintas maneras en que se pueden manejar las API’S.

Yo creo que el ejemplo es sencillo, como primer parámetro obtenemos, el handle del sub-menú en este caso el número 1, luego le pasamos el ítem en el cual queremos posicionar la imagen, en nuestro caso pusimos 0, colocamos MF_BYPOSITION, y luego colocamos la imagen del Picture1. Recomiendo que en ves de utilizar objetos Picture utilicen objetos ImageList. Yo lo utilice por cuestiones didácticas.

Las imágenes que coloquen en los menús, deben de tener un tamaño de 16x16 y 16 colores. Inclusive no he descubierto el por que, algunas veces las imágenes de 16x16 no encajan todas, como las imágenes de los programas como Corel, Office, etc. 

MODIFYMENU

Api: Private Declare Function ModifyMenu Lib "user32" Alias "ModifyMenuA" (ByVal hMenu As Long, ByVal nPosition As Long, ByVal wFlags As Long, ByVal wIDNewItem As Long, ByVal lpString As Any) As Long

Si apretaron el botón que dice “IMAGEN EN MENU 1” y luego se van al menú de “Archivo” verán que el texto se suplanto por imágenes. Esto lo logramos gracias a la API MODIFYMENU. Esta API lo que hace es cambiar estilo de un ítem del menú. En este caso cambia el estilo de Texto a Gráfico.

Como parámetros tenemos:

hMenu = Es el SUB-MENU en donde se encuentran los ítem a modificar.

nPosition = es el “ítem” al que queremos modificar.

wFlags = Aquí colocamos el estilo que queremos aplicar, en este caso es MF_BITMAP.

wIDNewItem = Es el número que identifica al nuevo ítem-. En nuestro caso usamos el mismo número que el anterior.

lpString = es una variable de tipo “Any”, para la cual le pasamos la imagen pero convirtiéndola a una variable de tipo “long” ya que es la única manera para que procese la imagen.

En el formulario lo tenemos de esta manera:

Private Sub Command2_Click()

    ModifyMenu GetSubMenu(GetMenu(Me.hwnd), 0), GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), 0), MF_BITMAP, GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), 0), CLng(ImageList1.ListImages(1).Picture)

    ModifyMenu GetSubMenu(GetMenu(Me.hwnd), 0), GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), 1), MF_BITMAP, GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), 1), CLng(ImageList1.ListImages(2).Picture)

    ModifyMenu GetSubMenu(GetMenu(Me.hwnd), 0), GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), 2), MF_BITMAP, GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), 2), CLng(ImageList1.ListImages(3).Picture)

End Sub

ModifyMenu  GetSubMenu(GetMenu(Me.hwnd), 0) “Obtenemos el handler del Sub-menú número 0”,  GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), 0) “Obtenemos el ítem número 0 que es el primero en modificarse”, MF_BITMAP “Cambiamos el estilo del ítem, a un estilo Gráfico”, GetMenuItemID(GetSubMenu(GetMenu(Me.hwnd), 0), “Asignamos el nuevo número de ITEM para el menú modificado, en nuestro caso mantenemos el mismo número”, CLng(ImageList1.ListImages(3).Picture), por último convertimos la imagen en tipo Long para que pueda ser procesada por la función.

POPMENU EN CONTROLES

En un forum un día un colega me pregunto si se podía quitar el POPMENU del Textbox el cual posee los comandos básicos para el manejo de texto, como copiar, pegar, cortar, etc. El decía que era imposible hacerlo. Y yo le dije que en la programación nada es imposible, pero si existen problemas complejos.

Si ustedes presionen el botón secundario en el Textbox y en el ListBox van a ver un POPMENU para cada uno. Esto lo hacemos empleando cierto “Truco” el cual consta en crear los menús en un formulario distinto al que estamos trabajando y luego mediante unas líneas de códigos sencilla, llamamos a los sub-menús del otro formulario que en nuestro caso es Form2 y lo mostramos en el control cuando aprietan el botón secundario.

Private Sub List1_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)

If Button = 2 Then

    Dim Puntero As POINTAPI

    GetCursorPos Puntero

    TrackPopupMenu GetSubMenu(GetMenu(Form2.hwnd), 1), TPM_LEFTALIGN, Puntero.X, Puntero.Y, 0, Form2.hwnd, ByVal 0&

End If

End Sub

Private Sub Text1_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)

If Button = 2 Then

    Dim Puntero As POINTAPI

    GetCursorPos Puntero

    TrackPopupMenu GetSubMenu(GetMenu(Form2.hwnd), 0), TPM_LEFTALIGN, Puntero.X, Puntero.Y, 0, Form2.hwnd, ByVal 0&

End If

End Sub

Los función API ya esta explicada y lo único que hay que aclarar es esta GetSubMenu(GetMenu(Form2.hwnd), 0), en la cual hacemos referencia al submenú número 0 pero del formulario Form2. Ustedes se preguntaran ¿Por qué en otro formulario y no en el mismo?. Estos submenús no son para mostrarlos al cliente en el menú principal sino como POPMENU en diferentes controles, si los colocas en el formulario principal, el cliente vería el menú, sin necesidad de presionar el botón secundario del mouse. Entonces para solventar el problema lo escondemos en un formulario que nunca lo va a ver el cliente, el cual posee unos menús que los mostraremos a nuestra conveniencia.

AREA NO CLIENTE 

Para finalizar con esta explicación del menú vamos a explicar este último trozo de código:

Private Sub Command1_Click()

If bool Then

    Dim Numero As Long

    hMenuSystem = GetSystemMenu(Me.hwnd, False)

    Numero = GetMenuItemCount(hMenuSystem)

    RemoveMenu hMenuSystem, Numero - 1, MF_BYPOSITION Or MF_DISABLED

    DrawMenuBar Me.hwnd

    bool = False

End If

End Sub

Este código se encarga de deshabilitar el botón “X” o cerrar del formulario. Ahora que ocurre en este código, empecemos explicando las API’S implicadas.

GETSYSTEMMENU

Api: Private Declare Function GetSystemMenu Lib "user32" Alias "GetSystemMenu" (ByVal hwnd As Long, ByVal bRevert As Long) As Long

Esta API nos retorna el menú del sistema, o en este caso el menú del formulario donde estamos trabajando. Ustedes dirán ¿QUÉ MENU?. Si ustedes corren el programa y le dan un click al icono superior izquierdo, el cual tiene la imagen del formulario, van ver un menú que sobresale que tiene como ITEMS, “RESTAURAR”, “MINIMIZAR”, ..., “ETC”, ese es el menú.

Como parámetros tenemos:

hwnd =  el handler de la ventana de la cual queremos obtener su System menú.-

bRevert = es un parámetro que se la pasa TRUE o FALSE, si es TRUE la función reestablece el menú por defecto, cualquier modificación que se le haya hecho se pierde.

Si es FALSE habilita el menú para que pueda ser modificado. Que en nuestro caso es lo que vamos hacer.

En nuestro código aparece de esta manera:

    hMenuSystem = GetSystemMenu(Me.hwnd, False)

GETMENUITEMCOUNT

Api: Private Declare Function GetMenuItemCount Lib "user32" (ByVal hMenu As Long) As Long

Bueno creo que el nombre habla por si solo, esta API nos retorna cuantos ITEM tiene un SUB-MENU. Como único parámetro se le pasa el handler del submenú en el cual trabajaremos.

Numero = GetMenuItemCount(hMenuSystem)

En nuestro caso utilizamos hMenuSystem que es nuestra variable que referencia al menú del sistema.

REMOVEMENU

Api: Private Declare Function RemoveMenu Lib "user32" (ByVal hMenu As Long, ByVal nPosition As Long, ByVal wFlags As Long) As Long

Esta API se utiliza para Remover una opción del menú que nosotros elijamos.

Como parámetros tenemos:

hMenu: es el menú al cual hacemos referencia.

nPosition: el número de ítem el cual queremos modificar.

wFlags: la bandera que determina que queremos hacer.


Const MF_DISABLED = &H2&

En nuestro código lo empleamos de esta manera:

RemoveMenu hMenuSystem, Numero - 1, MF_BYPOSITION Or MF_DISABLED

hMenuSystem ya saben que es!; Numero – 1 “Número es el la variable que contiene el número total de ITEM, si se dan cuenta cuando ustedes abran el menú, la opción “CERRAR” es la última, significa que si la variable “Número es 5”, la opción “Cerrar” será Numero – 1 ya que los menús se inician en 0”. MF_BYPOSITION Or MF_DISABLED, “Aquí procedemos a desactivar el menú con MF_DISABLED.”

Hasta aquí hemos deshabilitado la opción cerrar del menú. A ustedes les dejo la interpretación del ¿POR QUÉ USO UNA VARIABLE BOOL PARA ENTRAR UNA SOLA VEZ AL CODIGO? Pongan en comentario en la línea que contiene BOOL y aprieten el botón varias veces, y vayan viendo el menú a medida que lo presionan. Y entenderán por que puse esa variable.

Ahora ya deshabilitados tenemos que REDIBUJAR el MENU del sistema esto lo hacemos con el API siguiente.

DRAWMENUBAR

Api: Private Declare Function DrawMenuBar Lib "user32" (ByVal hwnd As Long) As Long

Bueno creo que esta muy sencilla de entender, lo que hace esta API  es de redibujar el menú Bar de un formulario. Por esa razón en el código se coloca de esta forma:

    DrawMenuBar Me.hwnd

Bueno con esto finalizo la explicación de las API’S en este ejemplo sobre MENUS, les dejo como pequeña tarea el ¿ESTANDO ELIMINADO EL BOTON “X” COMO HACEMOS PARA VOLVERLO  ACTIVAR?.

API’S INFORMATIVAS

En este tema vamos a tratar aquellas API’S que nos sirven para extraer información del Sistema Operativo. Debido a que son sencillas voy a mostrar como se declaran y sus parámetros principales.

GETWINDOWSDIRECTORY

Api: Private Declare Function GetWindowsDirectory Lib "kernel32" Alias "GetWindowsDirectoryA" (ByVal lpBuffer As String, ByVal nSize As Long) As Long

Esta API nos retorna la carpeta en donde fue instalado el sistema operativo Windows

Ejm:

Dim carpeta as String

Dim size as long

size = 255

Carpeta = Space$(size)

GetWindowsDirectory(carpeta, size)

GETSYSTEMDIRECTORY

Api: Private Declare Function GetSystemDirectory lib “kernel32” (byVal lpbuffer as String, ByVal nSize as long) as long

La misma que la anterior con la diferencia que retorna la dirección de la carpeta SYSTEM de Windows.

GETCOMPUTERNAME

Api: Private Declare Function GetComputerName Lib "kernel32" Alias "GetComputerNameA" (ByVal lpBuffer As String, nSize As Long) As Long

Retorna el nombre del equipo asignado en Windows.

GETUSERNAME

Api: Private Declare Function GetUserName Lib "advapi32.dll" Alias "GetUserNameA" (ByVal lpBuffer As String, nSize As Long) As Long

Retorna el nombre del usuario con el cual fue iniciado Windows.

EXITWINDOWSEX

Api: Private Declare Function ExitWindowsEx Lib "user32" (ByVal uFlags As Long, ByVal dwReserved As Long) As Long

Con esta API podemos apagar, resetear o suspender la computadora.

Valores de uFlags:

Public Const EWX_LOGOFF = 0

Public Const EWX_POWEROFF = &H00000008

Public Const EWX_REBOOT = 2

Public Const EWX_SHUTDOWN = 1

Public Const EWX_FORCE = 4

Public Const EWX_FORCEIFHUNG = &H00000010

El parámetro “dwReserved” se le coloca 0.

BUSCANDO VENTANAS

En los forum, otra de las preguntas que mas colocan es ¿Cómo cerrar la ventana INTERNET EXPLORER automáticamente?. Bueno señores, la respuesta es con el API “FindWindow”, esta API nos permite cerrar cualquier ventana abierta ya sea pasando como dato el  nombre de la ventana o su clase.

A continuación, coloquen este código en un formulario en VB, y antes de correrlo abran la “Calculadora” de Windows.

Private Declare Function FindWindow Lib "user32" Alias "FindWindowA" (ByVal lpClassName As String, ByVal lpWindowName As String) As Long

Private Declare Function SendMessage Lib "user32" Alias "SendMessageA" (ByVal hwnd As Long, ByVal wMsg As Long, ByVal wParam As Long, lParam As Any) As Long

Const WM_CLOSE = &H10

Const Word = "OpusApp"

Const Excel = "XLMAIN"

Const IExplorer = "IEFrame"

Const MSVBasic = "wndclass_desked_gsk"

Const NotePad = "Notepad"

Private Sub Form_Load()

    Whwnd = FindWindow(vbNullString, "Calculadora")

    If Whwnd = 0 Then

        MsgBox "Programa no existente"

    End If

    SendMessage Whwnd, WM_CLOSE, 0&, 0&

End Sub

Si corren el código, teniendo la “Calculadora” abierta, notarán que ya ejecutado el formulario la calculadora se cerró. Ahora como funciona FindWindow.

FINDWINDOW

Api: Private Declare Function FindWindow Lib "user32" Alias "FindWindowA" (ByVal lpClassName As String, ByVal lpWindowName As String) As Long

Esta API busca una ventana que se encuentre abierta, esta API retorna el HWND de la ventana encontrada. Si la ventana no es encontrada retorna 0.

Hay dos maneras, de buscar una ventana o usamos el parámetro “lpClassName”, o usamos “lpWindowName” que nos es mas que el título de la ventana, en el momento que hasta abierta.

    Whwnd = FindWindow(vbNullString, "Calculadora")

Esta línea lo que hace es BUSCAR una ventana llamada “Calculadora”, como verán, ese es el título de la Ventana de la calculadora. Al encontrarla la función me retorna su “hwnd” el cual almaceno en la variable “Whwnd”. Luego por SendMessage obligo a cerrar la ventana. 

Ahora la segunda manera es por “lpClassName” que es el nombre de la clase de la ventana, como verán en el código están estas líneas:

Const Word = "OpusApp"

Const Excel = "XLMAIN"

Const IExplorer = "IEFrame"

Const MSVBasic = "wndclass_desked_gsk"

Const NotePad = "Notepad"

Estos son los nombres de las clases, de los programas “Word”, “Excel”, etc. Por ejemplos si ustedes abren el Internet Explorer y suplantan la línea de FindWindow por esta:

    Whwnd = FindWindow(IExplorer, vbNullString)

En el momento que corran el programa el Internet Explorer se cierra automáticamente. Pueden notar lo siguiente o buscamos por el nombre de la clase o buscamos por el título de la ventana. Por esa razón notarán el como invierto el “vbNullstring” en ambos ejemplos.

CAMBIANDO LA RESOLUCIÓN DE WINDOWS

Aquí les presento las 2 Api’s necesarias para cambiar la resolución de Windows. Pära este ejemplo necesitan un formulario con un Listbox y un Boton.

Const CCDEVICENAME = 32

Const CCFORMNAME = 32

Private Type DEVMODE

    dmDeviceName As String * CCDEVICENAME

    dmSpecVersion As Integer

    dmDriverVersion As Integer

    dmSize As Integer

    dmDriverExtra As Integer

    dmFields As Long

    dmOrientation As Integer

    dmPaperSize As Integer

    dmPaperLength As Integer

    dmPaperWidth As Integer

    dmScale As Integer

    dmCopies As Integer

    dmDefaultSource As Integer

    dmPrintQuality As Integer

    dmColor As Integer

    dmDuplex As Integer

    dmYResolution As Integer

    dmTTOption As Integer

    dmCollate As Integer

    dmFormName As String * CCFORMNAME

    dmUnusedPadding As Integer

    dmBitsPerPel As Integer

    dmPelsWidth As Long

    dmPelsHeight As Long

    dmDisplayFlags As Long

    dmDisplayFrequency As Long

End Type

Const CDS_UPDATEREGISTRY = &H1

Const CDS_TEST = &H4

Const DISP_CHANGE_SUCCESSFUL = 0

Const DISP_CHANGE_RESTART = 1

Private Declare Function EnumDisplaySettings Lib "user32" Alias "EnumDisplaySettingsA" (ByVal lpszDeviceName As Long, ByVal iModeNum As Long, lpDevMode As Any) As Boolean

Private Declare Function ChangeDisplaySettings Lib "user32" Alias "ChangeDisplaySettingsA" (lpDevMode As Any, ByVal dwFlags As Long) As Long

Dim dev As DEVMODE

Dim i, resultado1, resultado2 As Long

Dim Resul As Boolean

Private Sub Command1_Click()

    dev.dmPelsWidth = Mid(List1.Text, 1, InStr(1, List1.Text, "X") - 2)

    dev.dmPelsHeight = Mid(List1.Text, InStr(1, List1.Text, "X") + 2, 3)

    dev.dmBitsPerPel = Mid(List1.Text, InStr(1, List1.Text, "X") + 5, InStr(1, List1.Text, "b") - (InStr(1, List1.Text, "X") + 5))

    resultado1 = ChangeDisplaySettings(dev, CDS_TEST)

    If resultado1 = DISP_CHANGE_RESTART Then

        MsgBox "Necesitas reiniciar la maquina"

    End If

    If resultado1 = DISP_CHANGE_SUCCESSFUL Then

        ChangeDisplaySettings dev, CDS_UPDATEREGISTRY

    End If

End Sub

Private Sub Form_Load()

    Command1.Caption = "Cambiar Resolucion"

    Do

        Resul = EnumDisplaySettings(0&, i, dev)

        List1.AddItem dev.dmPelsWidth & " X " & dev.dmPelsHeight & " " & dev.dmBitsPerPel & " bits"

        i = i + 1

    Loop While Resul

End Sub

Si ejecutan el programa verán que en el ListBox se colocan todas las Resoluciones posibles que soporta la tarjeta de video que posean, seleccionen una y denle click a “Cambiar Resolución” e inmediatamente verán como la computadora cambia de resolución. 

ENUMDISPLAYSETTINGS

Api: Private Declare Function EnumDisplaySettings Lib "user32" Alias "EnumDisplaySettingsA" (ByVal lpszDeviceName As Long, ByVal iModeNum As Long, lpDevMode As Any) As Boolean

Con esta API obtenemos información acerca de un dispositivo gráfico, en nuestro caso la usamos para hacer un listado de las resoluciones posibles a las cuales podemos cambiar.

Como parámetros tenemos 

lpszDeviceName: este parámetro puede ser NULL. Si es NULL nos referimos al dispositivo gráfico usado en ese momento.

iModeNum: sirve como índice, para recuperar la información del dispositivo gráfico.

lpDevMode: Es un puntero a una estructura de tipo DEVMODE la cual en los inicios de este documento se encuentra su explicación.

En el programa lo usamos de esta manera:

    Do

        Resul = EnumDisplaySettings(0&, i, dev)

        List1.AddItem dev.dmPelsWidth & " X " & dev.dmPelsHeight & " " & dev.dmBitsPerPel & " bits"

        i = i + 1

    Loop While Resul

End Sub

Como podrán observar, la estructura DEVMODE representada en la variable “dev”, cambia sus valores cada vez que entra, con este código logramos que “Mientras Resul sea true” continué mostrando información, cuando ya se haya extraído toda la información EnumDisplaySettings retorna FALSE y sale del bucle. Para moverse en los distintos tipos de información usamos la variable “i” la cual empieza en 0 y va aumentando de uno en uno.

CHANGEDISPLAYSETTINGS

Api: Private Declare Function ChangeDisplaySettings Lib "user32" Alias "ChangeDisplaySettingsA" (lpDevMode As Any, ByVal dwFlags As Long) As Long

Api utilizada para cambiar la configuración de un dispositivo gráfico. En nuestro caso lo usamos para cambiar la resolución del sistema operativo.

Como parámetros tenemos, “lpDevMode” el cual se le pasa una variable de tipo DEVMODE, y luego tenemos un Flag “dwFlag” el cual lo utilizamos para indicar lo que queremos hacer.

En el programa lo usamos de la manera siguiente:

    resultado1 = ChangeDisplaySettings(dev, CDS_TEST)

    If resultado1 = DISP_CHANGE_RESTART Then

        MsgBox "Necesitas reiniciar la maquina"

    End If

    If resultado1 = DISP_CHANGE_SUCCESSFUL Then

        ChangeDisplaySettings dev, CDS_UPDATEREGISTRY

    End If

Primero probamos si la resolución, a la cual estamos invocando es compatible, para ello, usamos un flag “CDS_TEST” si la función ChangeDisplaySettings retorna DISP_CHANGE_RESTART significa que para realizar el cambio de la resolución hay que reiniciar la maquina, si la función retorna DISP_CHANGE_SUCCESSFUL significa que no hay ningún problema. Ahora ¿Por qué invocar dos veces ChangeDsiplaySettings?

resultado1 = ChangeDisplaySettings(dev, CDS_TEST)

En este caso probamos si es compatible.

ChangeDisplaySettings dev, CDS_UPDATEREGISTRY

Y aquí sabiendo que ya es compatible, actualizamos esa información en el registro de Windows, y  para ello usamos el flag CDS_UPDATEREGISTRY

Aquí mostré lo básico de las dos funciones usadas para cambiar la resolución de un dispositivo gráfico. Si desean conocer, con detalle los valores que retornan, los flag, etc. Pueden consultarla en la ayuda de MSDN de Visual Studio.

CONCLUSION

Bueno damas y caballeros, hemos finalizado ya este manual de las API’S de WINDOWS en VB, solo les puedo decir que le he mostrado una pequeña porción de las API’S de Windows, ya que digamos que son APROX. unas 800 API’S y cuidado sino más, las que conforman al sistema operativo de Windows. Pero les aseguro para aquellas personas, que han visto interesante esta rama de la programación, que quienes conocen las API’S tiene el control del software que diseñan. Tal vez no tanto en el lenguaje de VB, pero si quieran entrar a un lenguaje de mas alto nivel como C++ o VC++, obligatoriamente tendrán que entender las API’S de Windows, ya que esos lenguajes giran entorno a ellas, muchos se preguntaran ¿Pero ya los controles de VB me simplifican el trabajo?. Y yo te respondería “Si tienes razón”, pero imagínate un problema en donde tu diseñes tu propios controles Adaptado a lo que tu quieres hacer, sin tener que establecer restricciones de ningún tipo. Entonces es ahí cuando necesitas las API’S.

Solo les puedo decir, como aprendiz de las API ya que a mi también me falta mucho camino por recorrer, que conocer las API’S de Windows o como le dicen algunos libros API Win32, es conocer a WINDOWS, si conoces a Windows te aseguro y tómame la palabra no habrá límites para el software que quieres diseñar. Ahora con eso no vayas a pensar que es ¡Fácil!, las API son complicadas y ahí cosas que aun todavía no entiendo, y hay problemas complicados que no se resolver, pero como le dije, a una muchacha en un forum que digo que era “Imposible hacer esto” y yo le respondí “Para la programación no hay nada imposible, pero si difícil”. Así que muchachos espero que estas páginas la hayan servido de algo y que les haya ayudado en su trabajo. 

Una cosa que quiero aclarar es que estas API’S son de Windows, pero las API no son exclusivas de Windows, existen también API’S Directx, API’S OpenGl, etc. Eso si, cuando vean API significa ¡trabajo fuerte!.

